code:
;-------------------------------------------------
;함수명 :FESTIVAL
;개요:행사를 나타내는 캐릭터라인을 취득하는 함수
;인수:없음
;-------------------------------------------------
@FESTIVAL
#FUNCTIONS
IF DAY:2 == 1
	SELECTCASE DAY:3
		CASE 1
			RETURNF "설날"
		CASE 3
			RETURNF "히나마츠리"
		CASE 11
			RETURNF "어버이날"
		CASE 14
			RETURNF "화이트 데이"
		CASE 25
			RETURNF "예대제"
	ENDSELECT
ELSEIF DAY:2 == 2
	SELECTCASE DAY:3
		CASE 7
			RETURNF "칠석"
		CASE 14
			RETURNF "추석"
	ENDSELECT
ELSEIF DAY:2 == 3
	SELECTCASE DAY:3
		CASE 1
			RETURNF "풍신 「입춘으로부터 210일째 되는 날」"
		CASE 15
			RETURNF "보름"
		CASE 27
			RETURNF "수확제"
		CASE 30
			RETURNF "할로윈"
	ENDSELECT
ELSEIF DAY:2 == 4
	SELECTCASE DAY:3
		CASE 3
			RETURNF "절분"
		CASE 14
			RETURNF "발렌타인 데이"
		CASE 24
			RETURNF "크리스마스 이브"
		CASE 25
			RETURNF "크리스마스"
		CASE 31
			RETURNF "그믐날"
	ENDSELECT
ENDIF

;특정의 축제의 기상시 메시지
@SEASONAL_MORNING
#DIM 생활시간보정, 5
SELECTCASE FESTIVAL()
	CASE "설날"
		FOR LOCAL,1,CHARANUM
			CFLAG:LOCAL:기상시간 = CSVCFLAG (LOCAL, GETNUM(CFLAG, "기상시간"))
			CFLAG:LOCAL:내방시간 = CSVCFLAG (LOCAL, GETNUM(CFLAG, "내방시간"))
			CFLAG:LOCAL:귀가시간 = CSVCFLAG (LOCAL, GETNUM(CFLAG, "귀가시간"))
			CFLAG:LOCAL:취침시간 = CSVCFLAG (LOCAL, GETNUM(CFLAG, "취침시간"))
		NEXT
		SIF DAY != 1
			CALL NEWYEAR
	CASE "발렌타인 데이"
		PRINTL 초콜릿을 받을 수 있을까…
	CASE "화이트 데이"
		FOR LOCAL, 1, CHARANUM
			SELECTCASE CFLAG:LOCAL:발렌타인
				CASE 1
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 블랙썬더를 주었군……
					PRINTFORMW 답례는 적당히 해도 될까
				CASE 2
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 보통 초콜릿을 주었군……
					PRINTFORMW 그만한 답례를 하자
				CASE 3
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 비싸 보이는 초콜릿을 주었군……
					PRINTFORMW 제대로 된 답례를 하자
				CASE 4
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 손수 만든 초콜릿을 주었군……
					PRINTFORMW 정성어린 답례를 하자
			ENDSELECT
		NEXT
	CASE "그믐날"
		PRINTL 오늘은 모든 캐릭터가 밤샘합니다
		FOR LOCAL,1,CHARANUM
			IF INRANGE(CFLAG:LOCAL:귀가시간, 900 , 1440) || INRANGE(CFLAG:LOCAL:귀가시간, 0 , 59) || INRANGE(CFLAG:LOCAL:취침시간, 900 , 1440) || INRANGE(CFLAG:LOCAL:취침시간, 0 , 119)
				생활시간보정:1 = 1500 - CFLAG:LOCAL:귀가시간
				생활시간보정:2 = 1560 - CFLAG:LOCAL:취침시간
				CFLAG:LOCAL:귀가시간 = TIMECIRCLE(CFLAG:LOCAL:귀가시간 + 생활시간보정:1)
				CFLAG:LOCAL:취침시간 = TIMECIRCLE(CFLAG:LOCAL:취침시간 + 생활시간보정:2)
			ENDIF
		NEXT
ENDSELECT

@TIMECIRCLE(ARG)
#FUNCTION
WHILE ARG > 1440
	ARG -= 1440
WEND
RETURNF ARG

@SEASONAL(ARG)
VARSET LOCAL
;만났을 때
SELECTCASE FESTIVAL()
	CASE "설날"
		PRINTL
		PRINTFORML ●%마스터는% %CALLNAME:ARG%에게 새해 인사를 했다
		IF TALENT:ARG:유치 && DAY > 1
			PRINTFORMW %CALLNAME:ARG%에게 세배돈을 졸라졌다
			SELECTCASE MONEY
				CASE IS > 500000
					LOCAL = 10000
				CASE IS > 100000
					LOCAL = 5000
				CASE IS > 30000
					LOCAL = 3000
				CASE IS > 3000
					LOCAL = 1000
				CASEELSE
					LOCAL = 0
			ENDSELECT
			IF !LOCAL
				PRINTFORML %마스터가% 지갑의 내용물을 보여주자, %조사처리(CALLNAME:ARG,"는")% 연민이 담긴 눈빛을 향했다
			ELSE
				PRINTFORML %CALLNAME:ARG%에게 \\{LOCAL}뺏겼다…
				PRINTFORML %조사처리(CALLNAME:ARG,"는")% 기분이 좋다
				MONEY -= LOCAL
				TALENT:ARG:기분 = 1
				CFLAG:ARG:호감도 += LOCAL/200
				CFLAG:ARG:신뢰도 += LOCAL/200
			ENDIF
		ENDIF
	CASE "발렌타인 데이"
		SELECTCASE CFLAG:ARG:호감도
			CASE 1000 TO 3000
				PRINTFORMW %조사처리(CALLNAME:ARG,"로")%부터 초콜릿을 받았다
				PRINTW ……블랙썬더다
				ITEM:120 ++
				CFLAG:ARG:발렌타인 = 1
			CASE 3001 TO 5000
				PRINTFORMW %조사처리(CALLNAME:ARG,"로")%부터 초콜릿을 받았다
				ITEM:121 ++
				CFLAG:ARG:발렌타인 = 2
			CASE IS > 5000
				IF EXP:ARG:애정경험 < 100
					PRINTFORMW %조사처리(CALLNAME:ARG,"로")%부터 비싸보이는 초콜릿을 받았다
					ITEM:122 ++
					CFLAG:ARG:발렌타인 = 3
				ELSE
					PRINTFORML %조사처리(CALLNAME:ARG,"로")%부터 초콜릿을 받았다
					PRINTW 손수 만든 것이다! 기쁘다!
					ITEM:123 ++
					CFLAG:ARG:발렌타인 = 4
				ENDIF
		ENDSELECT
ENDSELECT

@FESTIVAL_MENU()
#FUNCTIONS
SELECTCASE DAY:2
	;춘
	CASE 1
		SELECTCASE DAY:3
			CASE 1,2
				RETURNF "오세치요리"
		ENDSELECT
	;하
	CASE 2
	;추
	CASE 3
	;동
	CASE 4
		SELECTCASE DAY:3
			CASE 3
				RETURNF "에호마키"
			CASE 31
				RETURNF "토시코시 소바"
		ENDSELECT
ENDSELECT

@NEWYEAR
PRINTL 신년 새해복 많이 받으세요!
PRINTFORML %CALLNAME:MASTER%의 올해의 포부는…
PRINTL
PRINTLC [1] 자위 장려
PRINTLC [2] 동성애 추진
PRINTLC [3] 암컷 구멍 개척
PRINTL
PRINTL
PRINTLC [4] 항문 개발
PRINTLC [5] 유협진흥
PRINTLC [6] 자손 번영
$LOOP
INPUT
SELECTCASE RESULT
	CASE 1
		LOCALS = 자위
	CASE 2
		LOCALS = 동성애
	CASE 3
		LOCALS = 섹스
	CASE 4
		LOCALS = 애널 섹스
	CASE 5
		LOCALS = 파이즈리
	CASE 6
		LOCALS = 아이 만들기
	CASEELSE
		GOTO LOOP
ENDSELECT

PRINTFORML %마스터는% 올해 %LOCALS%에 전념하기로 굳게 마음먹었다
FLAG:포부 = RESULT

;연중 플래그를 리셋
FOR LOCAL, 0, CHARANUM
	CFLAG:LOCAL:연중 = 0
NEXT

path: ERB/이벤트관련/기념일이벤트_t.ERB


code:
@내방플래그
#DIM 내방자수

#DIM 내방확률
#DIM 기본내방확률
#DIM 내방인원수상한

;LOCAL = 각 캐릭터의 NO
내방자수 = 0
내방확률 = 0
기본내방확률 = 0
;출금인원수에 응해 기본 내방율의 조정. 실수가 아니고 비율 계산으로 변경하고 있습니다
SELECTCASE FLAG:출금인원수 * 100 / CHARANUM
CASE IS >= 60
	기본내방확률 = 15
CASE IS >= 40
	기본내방확률 = 16
CASE IS >= 20
	기본내방확률 = 17
CASEELSE
	기본내방확률 = 18
ENDSELECT

내방인원수상한 = 10
SIF TALENT:MASTER:천객만래
	내방인원수상한 = 12
;우선 확정으로 오는 캐릭터를 카운트
FOR LOCAL, 1, CHARANUM
	IF CFLAG:LOCAL:연회참가
		IF IN_HOME(FLAG:32)
			CFLAG:LOCAL:거점내방 = 1
			내방자수 ++
		;다만 신사외라면 확실히 오지 않는(술>>(넘을 수 없는 벽)>>연인의 당신)
		ELSE
			CFLAG:LOCAL:거점내방 = 0
		ENDIF
	;발원함 대상/의뢰실행중 및 달성 사람은 내방한다
	; // 자택 멤버는 발원함 플래그가 ON에서도 내방하지 않게 from 발원함 패치. ERB//
	ELSEIF (CFLAG:LOCAL:다음날내방 && !CFLAG:LOCAL:신사거주) || CFLAG:LOCAL:의뢰상황 == 3
		CFLAG:LOCAL:거점내방 = 1
		내방자수 ++
	ELSE
		CFLAG:LOCAL:거점내방 = 0
	ENDIF
	;와카사기히메의 내방플래그는 풀장이 없으면 접힌다
	IF LOCAL == 93 && CFLAG:LOCAL:거점내방 && FLAG:풀장 < 8
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	;교류 모드로 플래그 때
	ELSEIF FLAG:교류내방모드 == 1 && CFLAG:LOCAL:호감도 <= 0
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	ELSEIF FLAG:교류내방모드 == 2
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	ENDIF
NEXT

CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10 , 0 , CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	;자택 거주 캐릭터라면 조우하지 않는다
	IF CFLAG:LOCAL:신사거주 > 0
		CFLAG:LOCAL:거점내방 = 0
		CONTINUE
	ENDIF
	;출금의 사람은 제외
	IF CFLAG:LOCAL:출금 || TALENT:LOCAL:스침
		CFLAG:LOCAL:거점내방 = 0
		CONTINUE
	ENDIF
	SIF CFLAG:LOCAL:거점내방
		CONTINUE
	;출금인원수＋각 소질에 의한 수정으로부터 내방율을 산출해, 판정
	;일부 캐릭터에게는 추가로 조건이나 확률 변동이 존재
	;내방자수가 내방인원수상한을 넘으면(자) 스킵
	SIF 내방자수 > 내방인원수상한
		CONTINUE
	내방확률 = 기본내방확률 - TALENT:LOCAL:연모 * 2 - TALENT:LOCAL:사모 - CFLAG:LOCAL:(420 + MAIN_MAP)
	SIF CFLAG:LOCAL:딸딸이들킴
		내방확률 += 5
	SELECTCASE LOCAL
		;플랑, 님
		CASE 50, 66
			IF !RAND:(2 + 내방확률)
				CFLAG:LOCAL:거점내방 = 1
				내방자수 ++
			ELSE
				CFLAG:LOCAL:거점내방 = 0
			ENDIF
		;쳇
		CASE 54
			IF !RAND:(10 + 내방확률)
				CFLAG:LOCAL:거점내방 = 1
				내방자수 ++
			ELSE
				CFLAG:LOCAL:거점내방 = 0
			ENDIF
		;도 고담&나
		CASE 61, 62
			IF !RAND:(내방확률) && CFLAG:LOCAL:호감도 > 500
				CFLAG:LOCAL:거점내방 = 1
				내방자수 ++
			ELSE
				CFLAG:LOCAL:거점내방 = 0
			ENDIF
		;면월, 사구메는 예지상에는 가고 싶어하지 않는다
		CASE 98,99,110
			IF !TALENT:LOCAL:연모 && !TALENT:LOCAL:사모
				CFLAG:LOCAL:거점내방 = 0
			ELSEIF !RAND:(5 + 내방확률)
				CFLAG:LOCAL:거점내방 = 1
				내방자수 ++
			ELSE
				CFLAG:LOCAL:거점내방 = 0
			ENDIF
		CASEELSE
			IF !RAND:내방확률
				CFLAG:LOCAL:거점내방 = 1
				내방자수 ++
			ELSE
				CFLAG:LOCAL:거점내방 = 0
			ENDIF
	ENDSELECT

	;와카사기히메의 내방플래그는 풀장이 없으면 접힌다
	IF LOCAL == 93 && CFLAG:LOCAL:거점내방 && FLAG:풀장 < 8
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	;교류 모드로 플래그 때
	ELSEIF FLAG:교류내방모드 == 1 && CFLAG:LOCAL:호감도 <= 0
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	ELSEIF FLAG:교류내방모드 == 2
		CFLAG:LOCAL:거점내방 = 0
		내방자수 --
	ENDIF
NEXT

;외박중은 비표시
IF 내방자수 > 0 && AT_HOME(MASTER)
	PRINTFORMW %GET_MAPNAME(MAIN_MAP)%에 내방자가 있던 것 같다.
	PRINTL
ENDIF

license: mit


code:
@데이트종료타임업처리
VARSET LOCAL
;기력 제로, 쇠약시 데이트 중단
FOR LOCAL, 0, CHARANUM
	SIF CHK_DATENOW(CFLAG:LOCAL:데이트중) && (CFLAG:LOCAL:쇠약 || (BASE:LOCAL:기력 == 0 && CFLAG:LOCAL:현재위치 != 혼숙여관 && CFLAG:LOCAL:현재위치 != 러브호텔 && CFLAG:LOCAL:현재위치 != OMANEKIBEYA()))
		LOCAL:1 ++
	SIF CHK_DATENOW(CFLAG:LOCAL:데이트중) && CFLAG:LOCAL:쇠약
		CALL ENDUFUFU(LOCAL)
NEXT
SIF NEMUKE() >= 720 || BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
	LOCAL:2 ++
;초대받았으며 연속숙박이 아니라면 묵을 수 있다
IF CFLAG:MASTER:초대 && (LOCAL:1 || LOCAL:2) && !FLAG:70
	IF LOCAL:1
		PRINT 기력이 없어지기 시작했으나
	ELSEIF LOCAL:2
		PRINT 이만 돌아갈 시간이나
	ENDIF
	SIF !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 오늘은 %CALLNAME:(CFLAG:MASTER:초대)%가 묵게 해준다고 합니다
	SIF TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 오늘도 %CALLNAME:(CFLAG:MASTER:초대)%에 묵고 가겠습니까?
	PRINTL [0]묵고 간다 [1]돌아간다
	WHILE 1
	INPUT
	SIF RESULT == 0 || RESULT == 1
		BREAK
	WEND
	IF !RESULT && !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 호의를 받아들여 묵고 가기로 했다
		;데이트는 종료
		CFLAG:MASTER:데이트중 = MAIN_MAP
		CFLAG:TARGET:데이트중 = MAIN_MAP
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		CALL 모리야복권데이트상대의리셋()
		;이후의 처리를 컷
		GOTO OTOMARI
	ELSEIF !RESULT && TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 아무리 그래도 연달은 숙박은 거절당했다…
	ENDIF
ENDIF
IF LOCAL:1 && CFLAG:MASTER:현재위치
	PRINTL
	PRINTW ***** 기력이 0이 되었기에, 귀로에 접어들었습니다 *****
	CALL 귀가의시간경과
	FLAG:70 = 0
	RESETBGCOLOR
ELSEIF LOCAL:2
	PRINTL
	IF BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
		PRINTW ***** 만취했기 때문에, 갈짓자 걸음으로 귀로에 접어들었습니다 *****
		PRINTFORMW %GET_MAPNAME(MAIN_MAP, 0)%에 돌아온 %마스터는% 그 자리에서 쓰러졌습니다
	ELSE
		PRINTW ***** 졸음이 엄습했기에, 귀로에 접어들었습니다 *****
	ENDIF
	CALL 귀가의시간경과
ENDIF
IF LOCAL:1 || LOCAL:2
	FOR LOCAL, 0, CHARANUM
		IF LOCAL && CFLAG:LOCAL:쇠약
			IF !CFLAG:LOCAL:신사거주
				CFLAG:LOCAL:초기위치 = SUKIMA()
				CFLAG:LOCAL:현재위치 = SUKIMA()
				CFLAG:LOCAL:거점내방 = 0
				CALL KOJO_MESSAGE_SEND("EVENT", 20, LOCAL, 2,0)
				PRINTFORMW 녹초가 된 %CALLNAME:LOCAL%는 돌아갔습니다
			ELSE
				PRINTFORM %CALLNAME:LOCAL%는 %NAME_FROM_PLACE(CFLAG:LOCAL:초기위치)%
				PRINTW 에 돌아왔습니다
				SIF CFLAG:MASTER:초기위치 != CFLAG:LOCAL:초기위치
					LOCK:(CFLAG:LOCAL:초기위치) = 1
				CFLAG:LOCAL:현재위치 = CFLAG:LOCAL:초기위치
				CALL KOJO_MESSAGE_SEND("EVENT", 3,1, LOCAL, 0)
			ENDIF
		ENDIF
		CALL SET_DATE(99, LOCAL)
		IF CFLAG:LOCAL:초대
			CFLAG:LOCAL:초대 = 0
			IF LOCAL == 0
			ELSEIF CFLAG:LOCAL:신사거주
				CFLAG:LOCAL:초기위치 = CFLAG:LOCAL:신사거주
			ELSE
				CFLAG:LOCAL:초기위치 = SUKIMA()
			ENDIF
		ENDIF
		SIF CFLAG:LOCAL:동침중 > 0
			CFLAG:LOCAL:동침중 = 0
		SIF CFLAG:LOCAL:우후후
			CALL ENDUFUFU(LOCAL)
		;장난 OFF
		CALL SET_PRANK(0,LOCAL,1)

	NEXT
	CALL 모리야복권데이트상대의리셋()
ENDIF
IF ((CFLAG:MASTER:현재위치 % 100) % 10) == 0 && (CFLAG:MASTER:현재위치 % 100) == 0
	IF CFLAG:MASTER:현재위치 == 1100
		FOR LOCAL, 0, CHARANUM
			SIF CHK_DATENOW(CFLAG:LOCAL:데이트중)
				CFLAG:LOCAL:현재위치 -= 1100
		NEXT
	ENDIF
	SIF CHK_DATENOW(CFLAG:MASTER:데이트중)
		PRINTFORMW %DATENAME_PLACE()%에 도착했습니다
	FOR LOCAL, 0, CHARANUM
		SIF CHK_DATENOW(CFLAG:LOCAL:데이트중)
			CFLAG:LOCAL:현재위치 += 10
	NEXT
ENDIF
$OTOMARI

path: ERB/이벤트관련/데이트종료타입업처리_t.ERB


code:
@DAIRY_CONDITION_EV23
SIF GETBIT(FLAG_DAIRY_EV23, 1)
	RETURN -1
SIF ITEM:미라다의돌 > 0
	RETURN -1
SIF RAND:15 > 0
	RETURN -1
SIF MONEY < DAIRY_EV23_미라다_바닥시세 && !GETBIT(FLAG_DAIRY_EV23, 4)
	RETURN -1
CALL DAIRY_EV23
RETURN 1

@DAIRY_EV23
IF !GETBIT(FLAG_DAIRY_EV23, 0)
	SETBIT FLAG_DAIRY_EV23, 0
	PRINTFORML 하루를 마치고 이제 잘까하는 무렵에 현관 쪽에서 목소리가 들렸다
	PRINTFORMW 「실례하겠습니다～」
	PRINTFORML 이런 시간에 누구일까?
	PRINTFORMW %마스터는% 졸린 눈을 비비면서 응대에 나섰다
	PRINTFORML 「처음 뵙겠습니다～」
	PRINTFORML 「훌륭한 상품을 소개하러 왔습니다～」
	PRINTFORML 그곳에 있던 사람은 팔짝팔짝 움직이는 토끼귀가 사랑스러운 방문 판매원이었다
	CALL ASK_M("되쫓아 버린다", 1, "설명을 들어본다", 1)
	IF !RESULT
		SETBIT FLAG_DAIRY_EV23, 1
		PRINTFORML 방문 판매는 사절이다
		PRINTFORML 「그런ー, 적어도 이야기만이라도ー」
		PRINTFORMW 더욱더 물고 늘어지는 판매원을 무시하고 문단속을 확인하고 잤다
		RETURN
	ENDIF
	SETBIT FLAG_DAIRY_EV23, 2
	PRINTFORMW 흥미가 끌린 %마스터는% 상품 설명을 들어보기로 했다
	PRINTFORM 「갑작스럽지만 고객님,
	CALL COLORMESSAGE("엣찌", C_H_PINK, 0, 1)
	PRINTFORML 는 좋아하시나요?」
	PRINTFORMW 「좋아하시는군요～ 이 상품, 틀림없이 만족하실 수 있습니다」
	PRINTFORMW 「이번에 소개드리는 것은 이 상품!」
	PRINTFORM 「짜잔!
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORML 입니다!」
	PRINTFORMW 그렇게 말하면서 판매원이 꺼낸 것은 의심스러운 반짝이는 결정과도 같았다
	PRINTFORML 「환상향에는 매력적인 여성분들이 정말 많이 계시네요」
	PRINTFORML 「절벽위의 꽃무녀님에, 아름답고 경외한 여신님」
	PRINTFORMW 「건방요염한 대요괴에서부터, 순진무구한 요정님에 이르기까지, 더 힘차게 나아갑니다」
	PRINTFORMW 「……편안하고 친근한 요괴토끼 같은 것도 있지만요～ 에헤헤」
	PRINTFORM 「고객님도 신경이 쓰이는 그 여자아이와
	CALL COLORMESSAGE("그런～일", C_H_PINK, 0, 1)
	PRINTFORM 이나
	CALL COLORMESSAGE("이런～일", C_H_PINK, 0, 1)
	PRINTFORMW , 해보고 싶다고 생각한적이 있으시죠?」
	FONTBOLD
	PRINTFORM 「
	CALL COLORMESSAGE("연모", C_H_PINK, 0, 1)
	PRINTFORML 에 빠지게 하는 것도 좋지만, 일\@ GETBIT(TALENT:MASTER:성별, 1) ? 부 # 처\@다처는 힘들어요」
	PRINTFORM 「당신을 원해! 독차지하고 싶다! 그렇게 바라는 것은
	CALL COLORMESSAGE("사랑하는 소녀", C_H_PINK, 0, 1)
	PRINTFORMW 의 숙명인 것입니다!」
	FONTREGULAR
	PRINTFORM 「그 소녀도 연모, 이 소녀도 연모… …삼각, 사각, 사랑의 다각형이 닿는 끝은 질퍽질퍽한
	CALL COLORMESSAGE("정욕의 아수라장", C_D_RED, 0, 1)
	PRINTFORMW 입니다」
	PRINTFORM 「
	CALL COLORMESSAGE("초록 눈동자 그 사람", C_GREEN, 0, 1)
	PRINTFORML 이 아주 좋아하는 음식이네요～ 무섭다 무서워」
	PRINTFORM 「
	CALL COLORMESSAGE("질투로 얼룩진 서스펜스 극장", C_D_RED, 0, 1)
	PRINTFORM 은
	CALL COLORMESSAGE("꿈의 환상 하렘", C_H_PINK, 0, 1)
	PRINTFORMW 과는 거리가 멀어요」
	FONTBOLD
	PRINTFORML 「좀더 부담없이 섹스하고 싶어!」
	PRINTFORM 「이
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORMW 이 그 꿈을 확실하게 이뤄줍니다!!」
	FONTREGULAR
	PRINTFORM 「
	CALL COLORMESSAGE("연모", C_H_PINK, 0, 1)
	PRINTFORM 는 귀찮아, 하지만 노리고
	CALL COLORMESSAGE("섹프", C_RED, 0, 1)
	PRINTFORML 를 만드는 것은 어려운 일이죠～」
	PRINTFORML 「애초에 미움을 받고 있으면 화제에 오르지도 않습니다」
	;箸にも棒にもかからないとは、例えば何か、企画書や、提案、アイデアなどがあったときに、それらが、候補にすら上がらないというときに使われます。つまり、話題にもあがらない全く持って問題外であるという意味になります
	;대충 화제에도 올라오지(걸리지) 않는 다는 뜻
	PRINTFORM 「그렇다고, 잘 대해보면 이번엔 상대방이
	CALL COLORMESSAGE("사랑", C_H_PINK, 0, 1)
	PRINTFORMW 에 빠져버린다」
	PRINTFORML 「압니다, 알아요, 그 고민!」
	FONTBOLD
	PRINTFORM 「뚜렷한 신체만의 관계를 원하는데, 아무래도
	CALL COLORMESSAGE("신뢰도", C_YELLOW, 0, 1)
	PRINTFORML 가 너무 높아버린다!」
	PRINTFORM 「이
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORMW 이 그 고민을 확실하게 해결 해버립니다!!」
	PRINTFORM 「그런데, 이쪽의
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORML !」
	FONTREGULAR
	PRINTFORM 「한번 사용하면 성욕중추에 압력을 가해 여자아이 뇌속에 고객님의
	CALL COLORMESSAGE("성적인 매력", C_RED, 0, 1)
	PRINTFORMW 이 박힙니다!」
	PRINTFORM 「모습을 보면
	CALL COLORMESSAGE("두근!", C_RED, 0, 1)
	PRINTFORM 시선이 마주치면
	CALL COLORMESSAGE("두근!", C_RED, 0, 1)
	PRINTFORM 손이 닿으면
	CALL COLORMESSAGE("두근!", C_RED, 0, 1)
	PRINTFORM 더이상
	CALL COLORMESSAGE("두근 두근", C_RED, 0, 1)
	PRINTFORML 이 멈추지 않는다!」
	FONTBOLD
	PRINTFORM 「한층 더
	CALL COLORMESSAGE("신뢰도", C_YELLOW, 0, 1)
	PRINTFORM 의 상승을 억제하고, 단단한 그 소녀도
	CALL COLORMESSAGE("섹프", C_RED, 0, 1)
	PRINTFORMW 로 일직선!」
	PRINTFORMW 「이런 멋진 아이템, 이제 구할 수 밖에 없죠!?」
	FONTREGULAR
	PRINTFORML 「그렇게 되면 신경쓰이는 것은 가격 쪽입니다」
	PRINTFORM 「
	CALL COLORMESSAGE("좋은 것", C_YELLOW, 0, 1)
	PRINTFORM 은
	CALL COLORMESSAGE("좋은 값", C_YELLOW, 0, 1)
	PRINTFORML 이라는 것이, 세상의 상식, 사람의 상식, 환상향의 상식」
	PRINTFORM 「
	CALL COLORMESSAGE("가격 이상", C_M_GREEN, 0, 1)
	PRINTFORMW 이란 맛있는 이야기는 좀처럼 있을 수 없습니다」
	PRINTFORM 「실은 이
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORML 은 순도가 높은 매우 희소한 소재로 만들어졌고」
	PRINTFORML 「완벽한 마법사가 몇 명이나 힘을 합쳐 술을 건 매우 수고 많은 상품인 것입니다」
	PRINTFORM 「그래서, 통상의 판매 루트라면 가격은
	CALL COLORMESSAGE(@"\\{DAIRY_EV23_미라다_정가}", C_RED, 0, 1)
	PRINTFORMW 로 되어 있습니다만……」
	FONTBOLD
	PRINTFORMW 「이럴 수가!」
	PRINTFORM 「이번에 가격
	CALL COLORMESSAGE(@"\\{DAIRY_EV23_미라다_저가}", C_RED, 0, 1)
	PRINTFORMW 에 제공이 실현했습니다!」
	FONTREGULAR
	PRINTFORM 「상식을 벗어난 환상향만의
	CALL COLORMESSAGE("특별 가격", C_RED, 0, 1)
	PRINTFORML 입니다!!」
	PRINTFORM 「이런
	CALL COLORMESSAGE("파격적인 가격", C_RED, 0, 1)
	PRINTFORMW 상식을 벗어난 환상향만의 특별가격입니다!!」
	FONTBOLD
	PRINTFORML 「자, 고객님!」
	PRINTFORM 「충실한 성생활을 약속하는
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORMW !!!!」
	PRINTFORML 「하나, 어떠세요?!」
	FONTREGULAR
	PRINTFORML 　현재의 소지금:\\{MONEY}
	CALL ASK_M("흥미없다!", 1, "비싸!", 1, "샀다!", MONEY >= DAIRY_EV23_미라다_저가)
	SELECTCASE RESULT
	CASE 0
		$IRANAI
		PRINTFORML 「……그런가요」
		PRINTFORMW 「시간을 내주셔서 감사합니다!」
		PRINTFORML 「여기, 저희 회사 상품의 샘플입니다」
		PRINTFORMW 「괜찮으시면 사용해 주세요」
		;우콘, 로션, 콘돔 세트
		CALL GET_ITEM("우콘드링크", 1, 1)
		CALL GET_ITEM("로션", 1, 1)
		CALL GET_ITEM("콘돔", 1, 1)
		PRINTFORMW 「그럼 실례하겠습니다」
		PRINTFORMW 그 판매원은 마지막으로 정중히 절을 하고 떠났다
		RETURN
	CASE 1
		PRINTFORMW 「비, 비싸!?」
		PRINTFORML 「뭐ー, 그렇죠」
		PRINTFORM 「구매자들도
		CALL COLORMESSAGE("효과는 훌륭하지만 가격이 비싸다", C_YELLOW, 0, 1)
		PRINTFORMW 는 의견이 다수 있었습니다」
		PRINTFORM 「저희 회사 쪽에서도 이
		CALL COLORMESSAGE("훌륭한 상품", C_YELLOW, 0, 1)
		PRINTFORML 을 한 분이라도 더 보내드리고 싶다고 분발해서」
		PRINTFORM 「이번에,
		CALL COLORMESSAGE("저렴하게 제공", C_RED, 0, 1)
		PRINTFORML 된 기획이 완성되었습니다만. 으음～」
		PRINTFORMW 「내 재량으로 너무 제멋대로 해버리면 혼날까……」
		PRINTFORMW 「…………」
		PRINTFORMW 「……알았습니다」
		FONTBOLD
		PRINTFORM 「
		CALL COLORMESSAGE(@"통상 가격\\{DAIRY_EV23_미라다_정가}", C_RED, 0, 1)
		PRINTFORM 의
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML !!!」
		PRINTFORM 「여기에,
		CALL COLORMESSAGE("로션 3개 세트", C_YELLOW, 0, 1)
		PRINTFORMW 를 붙여!」
		PRINTFORMW 「게다가!」
		PRINTFORM 「이쪽의
		CALL COLORMESSAGE("콘돔 10개 세트", C_YELLOW, 0, 1)
		PRINTFORML 도 많이 주어야겠어요!」
		PRINTFORM 「이
		CALL COLORMESSAGE("미라다 안심 세트", C_D_PINK, 0, 1)
		PRINTFORM 있으면, 만약 내일
		CALL COLORMESSAGE("섹프", C_RED, 0, 1)
		PRINTFORMW 가 생겨도 괜찮아!」
		PRINTFORM 「가격은……
		CALL COLORMESSAGE(@"\\{DAIRY_EV23_미라다_바닥시세}", C_RED, 0, 1)
		PRINTFORML 에 제공이 됩니다! 이제 더 이상 싸지지 않을 거예요!」
		PRINTFORM 「자, 빨리
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORM 을 손에 넣어,
		CALL COLORMESSAGE("팡팡", C_RED, 0, 1)
		CALL HPH_PRINT(@"HPH")
		CALL COLORMESSAGE("앙앙", C_RED, 0, 1)
		CALL HPH_PRINT(@"HPH")
		PRINTFORMW  가버립시다!!」
		FONTREGULAR
		PRINTFORML 　현재의 소지금:\\{MONEY}
		CALL ASK_M("필요없어!", 1, "샀다!", 1)
		SIF !RESULT
			GOTO IRANAI
		MONEY -= DAIRY_EV23_미라다_바닥시세
		;러브 로션 3개 세트, 콘돔 10개 세트
		CALL GET_ITEM("로션", 3, 1)
		CALL GET_ITEM("콘돔", 10, 1)
	CASE 2
		MONEY -= DAIRY_EV23_미라다_저가
	ENDSELECT
	SETBIT FLAG_DAIRY_EV23, 3
	;여기서 미라다 입수
	CALL GET_ITEM("미라다의돌", 1, 1)
	FONTBOLD
	PRINTFORMW 「구매해 주셔서 감사합니다!」
	FONTREGULAR
	CALL ASK_M("졸리고 이제 잘까?", 1, "시험삼아 써볼까?", 1)
	IF !RESULT
		$START_SLEEP
		PRINTFORML 「그럼 실례하겠습니다. 또 사주신다면 기쁘겠습니다」
		PRINTFORMW 당신은 판매원을 배웅하자 기대에 부풀며 잠이 들었다
		RETURN
	ENDIF
	PRINTFORMW ……근데 이거 어떻게 써?
	PRINTFORMW 「이런 깜빡하고 있었네요.실례가 많았습니다」
	PRINTFORML 「사용할 상대방의 얼굴을 떠올리며, 손에 쥐고 힘을 주면……」
	PRINTFORMW 「희미하게 빛을 내기 시작하는 겁니다!」
	PRINTFORML 눈앞에 있는 판매원의 얼굴을 떠올리며 손에 쥐고 힘을 주자……돌은 희미하게 빛나기 시작했다
	PRINTFORMW 과연 이것으로 되는가?
	$USE_NOW
	ITEM:미라다의돌 --
	SETBIT FLAG_DAIRY_EV23, 4
	PRINTFORML 「오! 빠른 이용 감사합니다!」
	PRINTFORMW 「목적의 여성은 어떤 분입니까? 역시 대물 노린 건가요? 하쿠레이 무녀라든지! 스키마 요괴라든지!」
	PRINTFORMW 「……에? 나! ……아, 안됩니다～ 그런 일을 하시면 곤란합니다～」
	PRINTFORMW 「앗! 안된다고요! 젖가슴 만지면 안돼!」
	PRINTFORML 「고객님?! 치마에 얼굴을 들이대지 마세요!?」
	CALL HPH_PRINT("「싫어……더러운데 혀를 집어넣다니……아앙HPH……마시는 것도 안돼HPH」", "W")
	PRINTFORML 　·
	PRINTFORML 　·
	PRINTFORML 　·
	IF GETBIT(TALENT:MASTER:성별, 1) && !TALENT:MASTER:1
		PRINTFORMW 동정상실(%CALLNAME:MASTER%)
		TALENT:MASTER:1 = 1
	ENDIF
	CALL SET_FIRSTKISS_MASTER("심야 자기 방에서 미라다의 돌 방문 판매원을 돌로 발정 시킨 혼탁함에 바쳤다")
	PRINTFORMW
	PRINTFORML 「고객님도 참, 사실 이런 거 안 되니까요～?」
	PRINTFORMW 「저 화났거든요? 듣고 있어요?」
	PRINTFORMW 「정말이지! ……다른 고객님 댁도 돌아야 하는데 떠나고 싶지 않아지잖아요」
	PRINTFORMW 「……굉장히, 기분 좋았으니까요. 그, 용서해주지 않아도… …역시 안돼요」
	PRINTFORML 「한번 더 꼬셔주세요. 그러면 용서해 줄테니까요」
	IF GETBIT(TALENT:MASTER:성별, 1)
		PRINTFORML 「또 샤워도 빌려주세요. 정액 투성이고 냄새가 지독해요」
	ELSE
		PRINTFORML 「또 샤워도 빌려주세요. 애액 투성이고 냄새가 지독해요」
	ENDIF
	WAIT
	CALL HPH_PRINT("「그럼 실례하겠습니다. 또 사주신다면 기쁘겠습니다HPH」", "W")
	PRINTFORMW 방문 판매원 토낭과 섹프가 되었습니다
	RETURN
ELSE
	PRINTFORML 하루를 마치고 이제 잘까하는 무렵에 현관 쪽에서 목소리가 들렸다
	PRINTFORMW 「실례하겠습니다～」
	PRINTFORML 이런 시간에 누구일까?
	PRINTFORMW 당신은 졸린 눈을 비비며 응대에 나섰다
	IF GETBIT(FLAG_DAIRY_EV23, 4)
		CALL HPH_PRINT("「에헤헤 또 와버렸어요HPH」", "L")
	ELSE
		PRINTFORM 「매번 친숙한
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML  방문판매에 왔습니다～」
	ENDIF
	PRINTFORMW 그곳에 있던 사람은 팔짝팔짝 움직이는 토끼귀가 사랑스러운 방문판매원이었다
	PRINTFORML 「그 후 상태는 어떠세요～?」

	IF !GETBIT(FLAG_DAIRY_EV23, 3)
		PRINTFORM 「이번에야 말로
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML 의 훌륭함을 이해해주셨으면 해서, 실례했습니다!」
	ELSE
		PRINTFORM 「
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML 의 위력은 실감하실 수 있으실까요?」
		SIF GETBIT(FLAG_DAIRY_EV23, 4)
			CALL HPH_PRINT("「……제가 더 실감하고 있을지도 모르겠네요～ 에헤헤 HPH」", "L")
	ENDIF
	WAIT

	$MIRADA_2ND
	PRINTFORML 　현재의 소지금:\\{MONEY}
	CALL ASK_M("더 이상 오지 않아도 된다", 1, "필요 없다", 1, "사고 싶다", MONEY >= DAIRY_EV23_미라다_바닥시세, "밀어넘어뜨린다!", GETBIT(FLAG_DAIRY_EV23, 4))
	SELECTCASE RESULT
	CASE 0
		IF GETBIT(FLAG_DAIRY_EV23, 4)
			PRINTFORML 「……네?」
			PRINTFORML 「네!? 그건 좀 심하지 않아요!? 버려진 건가요! ?」
			CALL ASK_M("더 이상 오지 말아라", 1, "농담이야", 1)
			SIF !RESULT
				GOTO MOUKONAI
			PRINTFORML 「뭐야 놀래키지 마세요～」
			PRINTFORML 「사실은 어떻게 하고 싶은 거예요?」
			GOTO MIRADA_2ND
		ELSE
			$MOUKONAI
			SETBIT FLAG_DAIRY_EV23, 1
			PRINTFORML 「하아, 그렇습니까?」
			PRINTFORMW 방문 판매원은 풀이 죽은 얼굴을 하고 떠났다
		ENDIF
	CASE 1
		GOTO IRANAI
	CASE 2
		CALL GET_ITEM("미라다의돌", 1, 1)
		FONTBOLD
		PRINTFORMW 「구매해 주셔서 감사합니다!」
		FONTREGULAR
		IF MONEY < DAIRY_EV23_미라다_저가
			PRINTFORML 「……라니, 어라～？」
			PRINTFORM 「저～, 고객님? 이 상품 가격이
			CALL COLORMESSAGE(@"\\{DAIRY_EV23_미라다_저가}", C_RED, 0, 1)
			PRINTFORMW 이 되고 있지만……」
			PRINTFORML 「음, 어쩔 수 없네요」
			PRINTFORM 「이번에는
			CALL COLORMESSAGE(@"\\{DAIRY_EV23_미라다_바닥시세}", C_RED, 0, 1)
			PRINTFORML 에 팔아 드릴까요」
			PRINTFORMW 「저와 고객님 사이니까, 특별할텐데요?」
			MONEY -= DAIRY_EV23_미라다_바닥시세
		ELSE
			MONEY -= DAIRY_EV23_미라다_저가
		ENDIF
		IF GETBIT(FLAG_DAIRY_EV23, 4)
			PRINTFORMW 「고객님은 쇼핑만으로 만족하나요～？」
			CALL ASK_M("하지 않는다", 1, "한다", 1)
			IF !RESULT
				PRINTFORML 「어라 피곤 하셨어요?」
				PRINTFORMW 「그럼 오늘은 이만 실례할게요」
				PRINTFORMW 판매원 토낭은 아쉬운 듯 떠났다
				RETURN
			ELSE
				CALL HPH_PRINT(@"「후훗 그렇죠～HPH」", "L")
				PRINTFORMW 「……상냥하게 부탁드릴게요?」
				GOTO DO_H
			ENDIF
		ENDIF
		CALL ASK_M("졸리고 이제 잘까?", 1, "……이 돌을 판매원씨에게 써보자", 1)
		SIF !RESULT
			GOTO START_SLEEP
		PRINTFORML 눈앞에 있는 판매원의 얼굴을 떠올리며 손에 쥐고 힘을 주자……돌은 희미하게 빛나기 시작했다
		GOTO USE_NOW
	CASE 3
		PRINTDATAL
			DATA 「시ー러 범해진다～」
			;やーん、おかされる～
			DATA 「정말ー 고객님은 바로 그런일을 해요～」
			DATA 「우와!? 너무 달라붙는군요!」
			DATA 「일 때문에 온 거니까 상품을 사 주세요～」
			DATA 「네, 네. 엣찌는 이불 깔때까지 자제해요～」
		ENDDATA
		CALL ASK_M("입맞춤으로 입을 맞추게 한다", 1, "팁을 쥐어준다(\\3000)", MONEY >= 3000, "팁을 쥐어준다(10CM)", MONEY:2 >= 10)
		SELECTCASE RESULT
		CASE 0
			PRINTFORMW 「응ー！？」
			PRINTDATAW
				DATALIST
					DATA 「…………」
					DATA 「앗! 안돼요……아직 전혀 부족하니까요. 이봐요, 좀 더 쭈욱 해요」
				ENDLIST
				DATA 「또 그렇게 속이고……아, 갑자기 젖꼭지 만지기는 없다니까요. 약한거 알면서」
				DATA 「……능숙하시군요. 왠지 화가 나네요. 그 여유」
				;手慣れてるなぁ。なーんか腹が立ちますね。その余裕
				DATA 「……고객님, 토끼의 성욕을 우숩게 보고 있네요. 어떻게 되든 모르니까요!」
			ENDDATA
		CASE 1, 2
			SIF RESULT == 1
				MONEY -= 3000
			SIF RESULT == 2
				MONEY:2 -= 10
			SELECTCASE RAND:3
			CASE 0
				PRINTFORMW 「……후와아, 돈입니까. 뭐ー 뭐 받을건 받아두겠습니다만」
			CASE 1
				PRINTFORML 「우햐ー. 돈을 노리고 안겨있는게 아닌데요?」
				PRINTFORMW 「고객님의 테크닉이 너무 굉장한 게 나쁘니까요!」
			CASE 2
				PRINTFORML 「으응～. 직접 돈을 받는 것보다는 그」
				PRINTFORMW 「다음에 같이 점심 먹으면서……」
				PRINTFORMW 「……"소리가 작아서 안 들렸다"? ……오ー케이, 알겠습니다.얼른 합시다」
			ENDSELECT
		ENDSELECT
		$DO_H
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORMW
		[SKIPSTART]
		;생각 중
		SELECTCASE RAND:3
		CASE 0
			SIF !GETBIT(TALENT:MASTER:성별, 0)
				GOTO ERO_SCENE_BASIC
			SELECTCASE RAND:4
			CASE 0
				PRINTFORML ……후우
				CALL HPH_PRINT("「……후우 HPH」", "W")
				PRINTFORML 「아ー, 위응으로 가버립니다」
				PRINTFORMW 「다 냈으면 흐늘흐늘이군요?」
				PRINTFORMW 「후후. 에헤헤. 이 감촉도 독특하네요～. 말랑말랑 하고 부드럽고 재미있습니다」
				PRINTFORMW 인후 여기서 놀지 말아라
			CASE 1
				PRINTFORMW 「……꺄아악!」
				PRINTFORML 「어째서 고간으로부터 얼굴의 (분)편까지 정액 날아 옵니까～. 건강이 너무 좋습니다!이 에로 마신!」
				CALL HPH_PRINT("「\"굉장히 기분 좋았다\"가 아니어요, 바보 HPH」", "W")
				PRINTFORML 「이나다, 머리카락에 걸리고 있고～. 냄새 잡히지 않아요? 압니까?」
				PRINTFORMW 「아아, 이제(벌써)!이상한 일 말하지 말아 주세요!조금은 반성해라ー！！」
			CASE 2
				PRINTFORML 「이봐이봐」
				PRINTFORMW 「몇 번 말해야 알고 있습니까? 싱싱한은 안 되는 것입니까는……」
				PRINTFORML 「이봐, 고무 붙여드리겠습니다로부터」
				CALL HPH_PRINT("「……네, 좋아요. 와 주세요 HPH」", "W")
			CASE 3
				PRINTFORML 「언제나 생각합니다만, 고객님은 대단한 양을 내는군요」
				PRINTFORMW 「무엇인가 비결이라든지 있습니까?」
				PRINTFORMW 「나날의 청소와 팬티 모양? ……무엇입니까 그것? 의미 몰라요」
			ENDSELECT
		CASE 1
			SIF !GETBIT(TALENT:MASTER:성별, 1)
				GOTO ERO_SCENE_BASIC
			SELECTCASE GET_TALENTNAME(GETNUM(TALENT, "가슴사이즈"), TALENT:MASTER:가슴사이즈)
			CASE "거유", "폭유"
				PRINTFORML ……포근포근
				PRINTFORMW 「으응～, 고객님은 젖가슴 크지요」
				PRINTFORML ……포근포근
				CALL HPH_PRINT("「좋겠다, 계속 손대고 있었던 구 되어버리는 HPH」", "W")
			CASEELSE
				GOTO ERO_SCENE_BASIC
			ENDSELECT
		CASE 2
			GOTO ERO_SCENE_BASIC
			;어떻게든(어쩐지) 만들었지만 판정을 별로(호빵) 채우기라고 없다
			;판매원씨는 몸을 당긴다
			PRINTFORML 판매원과 같이 가 당신의 방까지 이동하려고 했지만, 거기에는 먼저 온 고객님이 있었다
			PRINTFORML
			PRINTFORML 「조금ー, 저것 누구입니까인가～？」
			PRINTFORML
			PRINTFORML 「……네? 자녀분?」
			PRINTFORML 「……네? 사모님?」
			PRINTFORML 「……네? 연인?」
			PRINTFORML
			PRINTFORML 「우와, 최저!바람둥이!여자의 적!」
			PRINTFORML 「나와 놀 때가 아니지 않습니까!」
			PRINTFORML
			PRINTFORML 「자자, 상관해 주지 않으면 외로워서 울음있어요?」
			PRINTFORML 「나에 대한 것은 좋으니까. 분명하게 귀여워해 주세요」
			PRINTFORML
			PRINTFORML 「나는 들키지 않는 동안에 실례합니다」
			PRINTFORML 「다음은 천천히」
			WAIT
			RETURN
		CASEELSE
			;특히 조건이 없는 패턴은 여기
			$ERO_SCENE_BASIC
			SELECTCASE RAND:5
			CASE 0
				PRINTFORML 「……그러니까요」
				PRINTFORML 「오늘은 급한 용무가 들어와 버려서. 곧바로 돌아가지 않으면 안된다  입니다」
				PRINTFORML 「이므로, 한번만 밖에 할 수 없습니다. 미안해요」
				PRINTFORML 「예, 이 장소에서. 선채로 합시다. 시간 없으니까」
				PRINTFORML 「1회 하면(자) 끝이에요?」
				PRINTFORML 「다음에 왔을 때는 충분히 상대 해 드릴테니까. 에에에, 빨리 해 주세요」
			CASE 1
				;미실장 분기
				;진한 에로 역시 쓸 수 없었다
				PRINTFORML 「오늘은이군요. 여기가 마지막 영업처입니다」
				PRINTFORML 「시간 만들려고 생각해 노력해 버렸습니다. 칭찬해 주세요. 에헤헤」
				PRINTFORML 「……많이 해요?」
				WAIT
				PRINTFORML 　·
				PRINTFORML 　·
				PRINTFORML 　·
				PRINTFORMW
				RETURN
			CASE 2
				PRINTFORML ……쪽
				PRINTFORMW 「고객님은 즐 있고입니다. 이 타이밍에 쪽을 하는 것은 반칙입니다」
				PRINTFORML ……쪽
				PRINTFORMW 「에헤헤. 답례♪」
				PRINTFORML ……쪽
				PRINTFORMW 「싫어!답례의 답례는 안 돼요」
				PRINTFORMW ……쭙, 쭙, 쭙, 쭙
			CASE 3
				PRINTFORML 「……아……아……싫어」
				PRINTFORMW 「유두, 약하니까……그런, 계속 스윽스윽 되면(자) ……」
				PRINTFORML 「으응 ……이상하게, 로부터……아」
				PRINTFORMW 「도, 부탁이니까……!……용서해!……괴롭히지 말고 창고 차이야 」
			CASE 4
				PRINTFORML 「!기다려!조금 기다려!」
				PRINTFORMW 「스톱!스～톱!」
				PRINTFORML 「어디 손대어 입니까!거기는 엉덩이의 구멍이에요!」
				PRINTFORMW 「그것은 안 돼요. 그런것은 안 돼요. 그런 일의 허가는 낼 수 없습니다」
				PRINTFORML 「보통으로 합시다. 보통이 좋아요」
				PRINTFORMW 「정말이지. 방심도 틈도 없네요, 고객님도 참」
			ENDSELECT
		ENDSELECT
		WAIT
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORMW
		[SKIPEND]
		PRINTFORML 「정신없는게 좀 아쉽네요」
		CALL HPH_PRINT(@"「고마웠습니다. 아주 좋았어요HPH」", "W")
		PRINTFORML 「언제가 될지는 모르겠지만 다시 올테니까요. 기다려 주세요」
		PRINTFORMW 「그럼 실례하겠습니다」
		PRINTFORMW 방문 판매원은 서운해하면서도 떠났다
	ENDSELECT
ENDIF

path: ERB/이벤트관련/데일리이벤트/DAILY_EV23 방문판매_t.ERB


code:
;DAIRY_EV23의 작성해 광역 변수를 갖고 싶어졌다
;다른 데일리용으로도 부디

;----------------------------
;EV23：방문판매
;----------------------------
#DIM SAVEDATA FLAG_DAIRY_EV23
;BIT0 이벤트가 한 번이라도 발생했다
;BIT1 맥 이루어, 이제 두 번 다시 판매원은 오지 않는다
;BIT2 설명을 끝까지 (들)물었다
;BIT3 미라다의 돌을 구입했다
;BIT4 판매원이 섹프가 되었다

#DIM CONST DAIRY_EV23_미라다_정가 = 980000
#DIM CONST DAIRY_EV23_미라다_저가 = 49800
#DIM CONST DAIRY_EV23_미라다_바닥시세 = 29800


code:
@DAIRY_CONDITION_EV0
SIF RAND:8 > 0
	RETURN -1
SIF DAY <= 30
	RETURN -1
SIF AVERAGE_YOGORE() > 800
	RETURN -1

CALL DAIRY_EV0
RETURN 1

@DAIRY_EV0
PRINTFORML %마스터가% 자는 사이에 %GET_MAPNAME(MAIN_MAP)%에 요정 대군이 몰려왔다!
IF RAND:10 < ABL:MASTER:전투능력 && AT_HOME(MASTER) == 1
	PRINTFORML 요정들은 %GET_MAPNAME(MAIN_MAP)% 안을 어지럽히고 있었지만
	PRINTFORML 도중에 눈치챈 %마스터가% 나오자마자, 쏜살같이 도망쳤다
	FOR LOCAL, MINROOM(), MAXROOM
		IF RAND:10 <= ABL:MASTER:전투능력
			YOGORE:LOCAL = MIN(YOGORE:LOCAL + 1000 + RAND:4000, 6000)
		ENDIF
	NEXT
;블랙썬더를 10개 이상 소지
ELSEIF ITEM:120 >= 10 && AT_HOME(MASTER) == 1
	;최대 9개(살). 당신의 몫은 남겨 둔다
	PRINTFORML %마스터가% 눈을 떠보니 요정들은 블랙썬더를 우물우물 하고 있었다
	PRINTFORML %마스터는% 요정들의 입 주위를 닦아주고 돌아가라고 타일렀다
	CALL LOST_ITEM("블랙썬더", RAND:9 + 1, 1)
ELSE
	PRINTFORML 요정들은 %GET_MAPNAME(MAIN_MAP)% 안을 어지럽혀 나갔다
	FOR LOCAL, MINROOM(), MAXROOM
		YOGORE:LOCAL = MIN(YOGORE:LOCAL + 1000 + RAND:4000, 6000)
	NEXT
ENDIF
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV0 어질러짐_t.ERB


code:
@DAIRY_CONDITION_EV10
#DIM YOME
#DIM CHILDNUM
SIF !CFLAG:MASTER:아이인원수
	RETURN -1
SIF RAND:20 > 0
	RETURN -1
YOME = 0
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	IF CFLAG:LOCAL:아이인원수 - CFLAG:LOCAL:육아인원수 > 0
		YOME = LOCAL
		CHILDNUM = RAND:(CFLAG:LOCAL:아이인원수 - CFLAG:LOCAL:육아인원수) + 1
		BREAK
	ENDIF
NEXT
IF YOME
	CALL DAIRY_EV10(YOME, CHILDNUM)
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV10(ARG, ARG:1)
;PRINTFORMW %CALLNAME:ARG%과의 아이의 %CHILDNAME:ARG:(ARG:1)%로부터 편지가 왔다
PRINTFORMW %조사처리(CHILDNAME:ARG:(ARG:1),"로")%부터 편지가 왔다
LOCAL:1 = 0
FOR LOCAL, 1, CHARANUM
	SIF CFLAG:LOCAL:호감도 > CFLAG:ARG:호감도
		LOCAL:1 ++
NEXT

TARGET = ARG
;말전용의 편지 이벤트 분기(없으면 통상 이벤트)
IF !RAND:2
	TRYCCALLFORM M_KOJO_EVENT_K{ARG}_CHILD_DAIRY_말편지(ARG:1)
	CATCH
		GOTO TEGAMI
	ENDCATCH
ELSE
	$TEGAMI
	IF LOCAL:1 >= 3
		TRYCCALLFORM M_KOJO_EVENT_K{ARG}_CHILD_DAIRY_공통편지(ARG:1, "바람둥이")
		CATCH
			PRINTFORMW 좀 더 %조사처리(CALLNAME:ARG,"를")% 소중히 하라고 적혀있었다……
		ENDCATCH
	ELSEIF !RAND:5
		TRYCCALLFORM M_KOJO_EVENT_K{ARG}_CHILD_DAIRY_공통편지(ARG:1, "염치없는돈") ;염치없는돈
		;金銭を無遠慮にねだって得ようとすること、思慮なく金をせびることを意味する語。親に小遣いをたかることを指す場合が多い。
		;용돈 달라는 뜻
		CATCH
			PRINTFORMW 용돈을 달라고 한다……
			;金の無心だ……
		ENDCATCH
		PRINTFORMW %마스터는% \@ MONEY > 30000? ￥30000을# 가진 돈 전부를\@ 보내기로 했다
		MONEY = MAX(MONEY - 30000,0)
	ELSE
		TRYCCALLFORM M_KOJO_EVENT_K{ARG}_CHILD_DAIRY_공통편지(ARG:1, "송금")
		CATCH
			PRINTFORMW %조사처리(CALLNAME:ARG,"와")% %마스터를% 신경쓰는 말과 송금\\10000이 들어가 있었다
		ENDCATCH
		MONEY += 10000
	ENDIF
ENDIF
TARGET = TARGET:1
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV10 아이로부터편지_t.ERB


code:
@DAIRY_CONDITION_EV11
IF !FLAG:자식기원
	RETURN -1
ELSEIF RAND:10 > 0
	RETURN -1
ELSE
	CALL DAIRY_EV11(FLAG:자식기원)
	RETURN 1
ENDIF

@DAIRY_EV11(ARG)
PRINTFORMW %조사처리(CALLNAME:ARG,"는")% %마스터와%의 아이가 생기도록 기도하고 있다……
CFLAG:ARG:자식보정 ++

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV11 자식기원_t.ERB


code:
@DAIRY_CONDITION_EV12
VARSET LOCAL
IF RAND:15 > 0
	RETURN -1
ELSE
	CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
	FOR LOCAL:10, 0, CHARANUM - 1
		LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
		IF CFLAG:LOCAL:신뢰도 >= 100 && CFLAG:LOCAL:신사거주
			LOCAL:1 = LOCAL
			BREAK
		ENDIF
	NEXT
	IF !LOCAL:1
		RETURN -1
	ELSE
		CALL DAIRY_EV12(LOCAL:1)
		RETURN 1
	ENDIF
ENDIF

@DAIRY_EV12(ARG)
#DIM 정답
#DIM 경험치
#DIM 제한시간
#DIM 배율
#DIM FEINT
#DIM 가짜
VARSET LOCAL
가짜 = 0
제한시간 = 2000 - (ABL:ARG:전투능력 - ABL:MASTER:전투능력) * 100
PRINTFORML %조사처리(CALLNAME:ARG,"로")%부터 신기술 훈련에 교제해달라는 말을 들었다
SIF ABL:ARG:전투능력 >= 4
		PRINTFORML 상대가 상대이므로 자칫 잘못하면 크게 다칠지도 모른다……
CALL ASK_YN("교제한다","싫다")
IF RESULT
	PRINTFORML %조사처리(CALLNAME:ARG,"는")% 아쉬운 것 같다
ELSE
	정답 = RAND:4
	경험치 = MAX(ABL:ARG:전투능력 * 5, 5)
	SELECTCASE RAND:20
		CASE IS <= 11
		;보통으로
			LOCALS:0 = 위
			LOCALS:1 = 오른쪽
			LOCALS:2 = 왼쪽
			LOCALS:3 = 아래
			배율 = 100
		CASE 12 TO 15
		;로마자
			LOCALS:0 = WI
			LOCALS:1 = OREUNJJOK
			LOCALS:2 = OENJJOK
			LOCALS:3 = ARAE
			배율 = 160
		;영어
		CASE 16,17,18
			LOCALS:0 = UP
			LOCALS:1 = RIGHT
			LOCALS:2 = LEFT
			LOCALS:3 = DOWN
			배율 = 120
		;스페인어
		CASE 19
			LOCALS:0 = encima de
			LOCALS:1 = a la derecha
			LOCALS:2 = a la izquierda
			LOCALS:3 = abajo de
			배율 = 300
	ENDSELECT
	;전투능력이 높으면 페인트 걸쳐 온다
	IF ABL:ARG:전투능력 > 3 && !RAND:3
		FEINT = 1
		제한시간 += 400
		배율 += 80
	ELSE
		FEINT = 0
	ENDIF
	CALL FISHER_YATES_SHAFFLE(4)
	PRINTFORMW %CALLNAME:ARG%의 공격을 피하자 (클릭으로 개시)
	PRINTFORM %CALLNAME:ARG%의
	SETCOLOR C_YELLOW
	IF FEINT
		PRINTFORM %조사처리(LOCALS:(RAND:4),"로")%
		RESETCOLOR
		PRINTFORM 속이고 %조사처리(LOCALS:(3 - 정답),"로")%
	ELSE
		PRINTFORM %조사처리(LOCALS:(3 - 정답),"로")%
		RESETCOLOR
	ENDIF
	PRINTFORML 부터의 공격!
	$LOOP_IKASAMA
	FOR LOCAL:10, 0,4
		LOCAL = SHAFFLE_ARRAY:(LOCAL:10)
		PRINTFORML [{LOCAL}] - %조사처리(LOCALS:LOCAL,"로")% 피한다
	NEXT
	SIF !가짜 && BASE:MASTER:TSP > 500
		PRINTFORML [{4}] - 시간정지
	TINPUT 제한시간, -1, 1
	IF RESULT == 정답
		PRINTFORML %마스터는% 공격을 회피했다!
		IF !가짜
			PRINTFORMW 서로에게 좋은 경험이 되었다
		ELSE
			PRINTFORMW 미끄러졌기 떄문에 그다지 연습이 되지 않았다……
		ENDIF
		경험치 = 경험치 * 배율 / 100
		MAXBASE:MASTER:체력+= 경험치
		EXP:MASTER:전투경험 += 경험치
		EXP:ARG:전투경험 += 경험치 * 2
		CFLAG:ARG:호감도 += 경험치 * 2
		CFLAG:ARG:신뢰도 += 경험치 * 2
		PRINTFORML 전투경험 ＋{경험치}(%CALLNAME:MASTER%)
		PRINTFORML 전투경험 ＋{경험치 * 2}(%CALLNAME:ARG%)
		PRINTFORML 호감도 ＋{경험치 * 2}(%CALLNAME:ARG%)
		PRINTFORML 신뢰도 ＋{경험치 * 2}(%CALLNAME:ARG%)
	ELSEIF RESULT == 4 && !가짜 && BASE:MASTER:TSP > 500
		가짜 = 1
		배율 /= 5
		PRINTFORMW %마스터는% 시점 정지 능력으로 공격을 판별했다!
		GOTO LOOP_IKASAMA
	ELSE
		PRINTFORML 피하지 못했다!
		PRINTFORM %마스터는%
		SELECTCASE ABL:ARG:전투능력
			CASE IS <= 1
				PRINTFORM 조금
			CASE 2,3
				PRINTFORM 상당히
			CASEELSE
				PRINTFORM 엉망진창
		ENDSELECT
		PRINTFORMW 아프다는 생각이 들었다
		BASE:MASTER:체력 -= 400 * (1 + ABL:ARG:전투능력)
		PRINTFORML 전투경험 ＋{경험치}(%CALLNAME:ARG%)
		EXP:ARG:전투경험 += 경험치
		CFLAG:ARG:호감도 += 경험치 / 2
		CFLAG:ARG:신뢰도 += 경험치 / 2
		PRINTFORML 호감도 ＋{경험치}(%CALLNAME:ARG%)
		PRINTFORML 신뢰도 ＋{경험치}(%CALLNAME:ARG%)
	ENDIF
ENDIF
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV12 특훈_t.ERB


code:
@DAIRY_CONDITION_EV13
#DIM 대상

SIF RAND:(10 - TALENT:MASTER:정욕감지)> 0
	RETURN -1
대상 = 0
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	SIF !CFLAG:LOCAL:신사거주
		CONTINUE
	IF CFLAG:LOCAL:욕구불만도 >  700
		대상 = LOCAL
		BREAK
	ENDIF
NEXT
SIF !대상
	RETURN -1

CALL DAIRY_EV13(대상)
RETURN 1

@DAIRY_EV13(ARG)
PRINTFORMW 한밤중 잠에서 깨버리고 말았다
PRINTFORMW 더 자려고 생각했지만 아무래도 잠들 수 있을 것 같지 않다……
CALL ASK_YN("돌아다닌다","방에 있는다")
IF RESULT
	PRINTFORM %마스터는% 졸려질 때까지
	PRINTDATA
		DATAFORM 콜렉션을 정리·감상하며 보냈다
		DATAFORM 미와 선, 팬티에 대해 생각을 깊이 했다
		DATAFORM 벽의 얼룩과 담소했다
	ENDDATA
	PRINTFORML ……
	CALL COLORMESSAGE("마음이 풍부해지면서 일시적으로 기력과 TSP가 좀 늘었다", C_YELLOW, 2,1)
	CALL BUFF_BASE(MASTER, BASE_기력, 100)
	CALL BUFF_BASE(MASTER, BASE_TSP, 100)
ELSEIF !RAND:4 || (!RAND:4 && ABL:ARG:친밀 >= 7)
	PRINTFORMW 적당히 돌아다니다가, \@ CFLAG:MASTER:311 == CFLAG:ARG:311 ? 자기 방으로 돌아왔더니 뭔가 소리가 들렸다# 이제 슬슬 돌아갈까 하다가 %CALLNAME:ARG%의 방에서 뭔가 들려서 들러봤다\@
	CALL ONABARE(ARG)
	CFLAG:MASTER:철야 = 1
ELSEIF ITEM:토부「개운대문」
		PRINTL 이런, 개운대문의 모습이…?
		CALL ASK_YN("꺼낸다","방치한다")
		IF !RESULT
			ITEM:토부「개운대문」 --
			SETCOLOR COLOR ("YELLOW")
			PRINTFORMW 토부「개운대문」이 발동!
			RESETCOLOR
			PRINTFORMW 적당히 돌아다니다가, \@ CFLAG:MASTER:311 == CFLAG:ARG:311 ? 자기 방으로 돌아왔더니 뭔가 소리가 들렸다# 이제 슬슬 돌아갈까 하다가 %CALLNAME:ARG%의 방에서 뭔가 들려서 들러봤다\@
			CALL ONABARE(ARG)
			CFLAG:MASTER:철야 = 1
		ELSEIF !RAND:2
			GOTO OYATU
		ELSE
			GOTO NANIMONAI
		ENDIF
ELSEIF !RAND:2
	$OYATU
	PRINTFORML 배가 조금 고프다 게다가 뭔가 먹으면 졸려질지도 모른다
	PRINTFORMW %마스터는% 부엌에 가서 찬장을 뒤지자, 맛있는 간식을 발견했다
	PRINTFORMW %마스터가% 산 것은 아니다. 그러나 누군가가 %마스터를% 위해 준비해줬을 가능성도 버리지 못한다
	PRINTFORML
	CALL COLORMESSAGE("어떠케, 이제 곧 유통기하니 지나! 누가 빨리 머거주면 안 될까!", C_YELLOW, 2,2)
	;ﾄﾞｳｼﾖｳ､ﾓｳｽｸﾞｼｮｳﾐｷｹﾞﾝｶﾞｷﾁｬｳﾜ!ﾀﾞﾚｶﾊﾔｸﾀﾍﾞﾃｸﾚﾅｲｶｼﾗ!
	;どうしょう､もうすぐしょうみきげんがきちゃうわ!だれかはやくたべてくれないかしら!
	;어떻게, 이제 곧 유통기한이 지나! 누가 빨리 먹어주면 안될까!
	PRINTFORML
	PRINTFORML 간식도 이렇게 말한다. 고맙게 받기로 하자
	CALL BUFF_BASE(MASTER, BASE_기력, 300)
	CALL COLORMESSAGE("맛있는 간식을 먹어 일시적으로 기력이 증가한다", C_YELLOW, 2,1)
	PRINTFORMW %마스터는% 이를 다시 닦고 기분 좋게 잠자리에 들었다
	CALL COLORMESSAGE(@"왠지 %CALLNAME:ARG%의 기분이 나빠진다……", C_RED, 2,1)
	TALENT:ARG:기분 = -1
ELSE
	$NANIMONAI
	PRINTFORMW 돌아다녀보긴 했지만 딱히 재미있는 일도 없고 얌전하게 다시 자기로 했다
	CFLAG:MASTER:취침시간 += 60
	CALL COLORMESSAGE("수면시간이 1시간 줄어들었다", C_YELLOW, 2,1)
ENDIF

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV13 자택에서_딸딸이들킴_t.ERB


code:
@DAIRY_CONDITION_EV15
SIF RAND:20 > 0 || FLAG:이상기상
	RETURN -1
CALL DAIRY_EV15
RETURN 1

@DAIRY_EV15
FLAG:이상기상 = 6 + RAND:2

TIME:5 = 0
SELECTCASE FLAG:이상기상
	CASE 6
		LOCALS = 치마가 넘어가 머리까지 닿으면 기분 좋아지게
	CASE 7
		LOCALS = 바람이 팬티를 싣고 오게
;	CASE 8 ;일판 4.755에서 DAIRY_EV26 으로 이관.
;		LOCALS = 분홍빛 바람이 불게
ENDSELECT

PRINTFORMW 지구온난화, 환경오염, 난민 문제, 대통령 선거, 팬더믹, 장시간 노동, 보행 중 폰질, Switch 품귀……
PRINTFORMW 바깥세상의 갖가지 문제는 환상향에도 영향을 미치지 않을 수는 없었다
CALL COLORMESSAGE(@"바깥세상의 영향으로 발생한 이상기상 때문에 %LOCALS% 되었다", C_PINK, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV15 이상기상_t.ERB


code:
@DAIRY_CONDITION_EV16
SIF !TFLAG:치마걷어올리기지나고
	RETURN -1
SIF RAND:(TFLAG:치마걷어올리기지나고) < RAND:300
	RETURN -1
CALL DAIRY_EV16
RETURN 1

@DAIRY_EV16
PRINTFORML 붕붕마루 신문의 호외가 나왔습니다!
PRINTFORMW 치마를 넘기는 %CALLNAME:MASTER%의 기사가 실려 있습니다!
FOR LOCAL, 1, CHARANUM
	CFLAG:LOCAL:호감도 = MAX(CFLAG:LOCAL:호감도 - 100, 0)
	CFLAG:LOCAL:신뢰도 = MAX(CFLAG:LOCAL:신뢰도 - 50, 0)
NEXT
CALL COLORMESSAGE("전원의 호감도가 100, 신뢰도가 50 떨어졌다……", C_YELLOW, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV16 치마걷어올리기_t.ERB


code:
@DAIRY_CONDITION_EV17
SIF RAND:20 > 0 || FLAG:이상기상
	RETURN -1
CALL DAIRY_EV17
RETURN 1

@DAIRY_EV17
FLAG:이상기상 = RAND:5 + 1
TIME:5 = 0
SELECTCASE FLAG:이상기상
	CASE 1
		LOCALS = 모두의 클리토리스가 민감
	CASE 2
		LOCALS = 모두의 여성기가 민감
	CASE 3
		LOCALS = 모두의 항문이 민감
	CASE 4
		LOCALS = 모두의 가슴이 민감
	CASE 5
		LOCALS = 모두의 입술이 민감
ENDSELECT

PRINTFORMW 영원정의 조약실이 폭발 사고를 냈습니다!
PRINTFORMW 기화한 약품은 캇파 공방의 매연과 반응해, 미지의 화합물이 되어 대기 중에 퍼져나간다……
CALL COLORMESSAGE(@"%LOCALS%해졌다", C_PINK, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV17 폭발사고_t.ERB


code:
@DAIRY_CONDITION_EV18
SIF RAND:15 > 0
	RETURN -1
CALL DAIRY_EV18
RETURN 1

@DAIRY_EV18
PRINTFORMW 달에서 요염한 빛이 쏟아져 내린다……

FOR LOCAL, 0, CHARANUM
	SIF ((TALENT:LOCAL:150 || TALENT:LOCAL:151) && !TALENT:LOCAL:초경) || !GETBIT(TALENT:LOCAL:성별, 0)
		CONTINUE
	SIF RAND:5 > 0
		CONTINUE
	CALL COLORMESSAGE(@"%CALLNAME:LOCAL%의 생리주기가 흐트러졌다", C_YELLOW, 1)
	CFLAG:LOCAL:900 = RAND:10 + 1
NEXT
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV18 생리주기랜덤화_t.ERB


code:
@DAIRY_CONDITION_EV19
SIF RAND:10 > 0
	RETURN -1
SIF FLAG:이상기상
	RETURN -1
;여름과 겨울만
SIF !GROUPMATCH(DAY:2, 2,4)
	RETURN -1
CALL DAIRY_EV19
RETURN 1

@DAIRY_EV19
FLAG:이상기상 = RAND:2 + 9

SELECTCASE DAY:2
	CASE 2
		FLAG:이상기상 = 9
		TIME:5 = 5
		PRINTFORML 바람이 엄청 분다……
	CASE 4
		FLAG:이상기상 = 10
		TIME:5 = 9
		PRINTFORML 오늘밤은 냉기가 심하다……
ENDSELECT

CALL COLORMESSAGE(@"내일은 %조사처리(GET_WEATHER(),"가")% 될 것 같다……", C_YELLOW, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV19 계절한정날씨_t.ERB


code:
@DAIRY_CONDITION_EV1
;영원정거주지라면1/10, 그 이외로1/30로 발생
SIF MAIN_MAP == 4 && RAND:10 > 0
	RETURN -1
SIF MAIN_MAP != 4 && RAND:30 > 0
	RETURN -1
CALL DAIRY_EV1
RETURN 1

@DAIRY_EV1
;여기에 이벤트 내용을 쓰자 최후는 RETURN 1해
IF MAIN_MAP != 4
	PRINTFORML 영원정에서 약 시공품이 도착했다
ELSE
	PRINTFORML 에이린에게 약 시공품을 받았다
ENDIF
CALL GET_ITEM("영양드링크", 1, 2)
RETURN 1


code:
@DAIRY_CONDITION_EV20
SIF RAND:15 > 0
	RETURN -1
SIF CFLAG:[[반키]]:신사거주
	RETURN -1
SIF CFLAG:[[반키]]:출금
	RETURN -1
SIF !TALENT:[[반키]]:사모 && !TALENT:[[반키]]:연모
	RETURN -1
CALL DAIRY_EV20
RETURN 1

@DAIRY_EV20
#DIM 손님수

PRINTFORML 한밤중, 창가에서 목소리가 들렸다
PRINTFORMW %마스터가% 졸린 눈을 비비며 창문을 열자, 머리가 떠 있었다
PRINTFORML ……세키반키의 머리다
PRINTFORML 지나가는 길에 들러봤다고 한다
;通りがかったから寄ってみたという
CALL ASK_YN("들어오게 한다","졸리다")
IF RESULT
	PRINTFORMW 졸리기 때문에 되돌려 보냈다
	RETURN
ENDIF
PRINTFORML 들어오도록 했다

손님수 = 1
IF ABL:[[반키]]:친밀 >= 10 && !RAND:3
	PRINTFORML 창문으로 머리가 들어왔다… …하지만, 하나뿐이 아니었다
	FOR LOCAL, 2, ABL:73:친밀 - 5
		SIF !RAND:2
			CONTINUE
		손님수 ++
		PRINTFORMW {손님수}번째 머리가 들어왔다
		SIF 손님수 >= 9
			BREAK
	NEXT
	SIF 손님수 == 1
		PRINTFORML 라고 생각했지만, 역시 하나였다
	SIF 손님수 >= 4
		PRINTFORMW 이만큼 같은 얼굴에 둘러싸지면, 역시 지내기가 불편하다……
ENDIF
IF TIME:5 >= 4
	PRINTFORML 젖어있었기 때문에 목욕타올을 건네주었다
	PRINTFORML 타올 위에서 굴러다니는 머리는 꽤 재미있는 광경이다
ELSE
	SELECTCASE RAND:3
		CASE 0
			PRINTFORML 눈높이를 맞추기 위해 책상 위에 방석을 놓고 앉도록 했다
			SIF 손님수 > 1
				PRINTFORML 책상 위에 생목이 늘어서 있는 광경은 마치 베어진 적군의 머리 수를 확인하는 것 같았다
				;獄門か首実検だ
		CASE 1
			PRINTFORML 차를 끓였더니 마시기 힘들어해서 빨대를 꽂았다
			PRINTFORML 마신 것은 도대체 어디로 가는 것일까……
		CASE 2
			PRINTFORML 뒷머리에 잎사귀가 걸려 있어서 때주었다
	ENDSELECT
ENDIF
PRINTFORMW
PRINTFORMW …
PRINTFORMW ……
PRINTFORMW ………

IF RAND:100 < ABL:[[반키]]:친밀 * 3 + TALENT:[[반키]]:연모 * 15
	PRINTFORM 잠시 담소한 후, 머리
	IF 손님수 > 1
	PRINTFORM 들은
	ELSE
	PRINTFORM 는
	ENDIF
	PRINTFORM 내일은 몸통째 온다며
	IF !PRIVATEROOM:(CFLAG:MASTER:초기위치 % 100) && TALENT:[[반키]]:연모
		PRINTFORML 돌아가려고 했지만 어차피 내일 올거면 자고 가라고 했다
		PRINTFORM %마스터는% 머리맡에 머리를
		IF 손님수 > 1
		PRINTFORM 나란히
		ENDIF
		PRINTFORML 놓고 잠들었다……
	ELSE
		PRINTFORML 돌아갔다
	ENDIF
	CFLAG:[[반키]]:다음날내방 = 1
ELSE
	PRINTFORM 잠시 담소한 후, 머리
	IF 손님수 > 1
	PRINTFORM 들은
	ELSE
	PRINTFORM 는
	ENDIF
	PRINTFORML 창문으로 돌아갔다
	SIF !RAND:2
		CFLAG:[[반키]]:다음날내방 = 1
ENDIF
;시스템 메시지 다하고 호감도 변화
;마지막 인수만 호감도가 변화
CALL CHANGE_CFLAG(2,[[반키]], 50)

;시스템 메시지 다하고 신뢰도 변화
;마지막 인수만 신뢰도가 변화
CALL CHANGE_CFLAG(4,[[반키]], 20)

;수면시간 감소
CFLAG:MASTER:취침시간 += 60
CALL COLORMESSAGE("수면시간이 1시간 줄어들었다", C_YELLOW, 2)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV20 생목님_어서오세요_t.ERB


code:
@DAIRY_CONDITION_EV21
#DIM YOME
SIF !FLAG:자식기원
	RETURN -1
SIF FLAG:획득팬티총수 < 100
	RETURN -1
YOME = FLAG:자식기원

SIF RAND:30 > 0
	RETURN -1
SIF TALENT:YOME:임신
	RETURN -1
SIF !TALENT:YOME:비생물 && !TALENT:YOME:봉래인 && !TALENT:YOME:신령
	RETURN -1
SIF CFLAG:YOME:아이인원수 > 0
	RETURN -1
CALL DAIRY_EV21(YOME)

RETURN 1

@DAIRY_EV21(YOME)
#DIM YOME
PRINTFORMW 팬티님은 자식 없는 %조사처리(CALLNAME:YOME,"를")% 불쌍히 여겼다……
CALL BUFF_BASE(MASTER, BASE_체력, 1000)
CALL BUFF_BASE(MASTER, BASE_기력, 1000)
CALL COLORMESSAGE(@"%CALLNAME:YOME%의 체력과 기력이 일시적으로 크게 강화되었다", C_YELLOW, 1,1)
CALL BUFF_BASE(MASTER, BASE_정력, 500)
CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 정력이 일시적으로 강화되었다", C_YELLOW, 1,1)
IF !TALENT:MASTER:농후정액
	TALENT:MASTER:농후정액 = 1
	CLEARBIT TFLAG:일일일회, 8
	CALL COLORMESSAGE(@"%마스터는%【농후정액】이 되었다", C_YELLOW, 1,1)
	BASE:MASTER:정력 = MAXBASE:MASTER:정력
ENDIF
PRINTW
CFLAG:YOME:다음날내방 = 1
CFLAG:YOME:자식보정 ++


code:
@DAIRY_CONDITION_EV22
#DIM YOME
#DIM CHILDNUM
SIF CMATCH(TALENT:육아중, 0) == CHARANUM
	RETURN -1
SIF RAND:5 > 0
	RETURN -1
YOME = 0
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	IF CHK_HAVE_CHILD(LOCAL, CHILD_취학, CHILD_자립) > 0
		YOME = LOCAL
		CHILDNUM = CHK_HAVE_CHILD(LOCAL, CHILD_취학, CHILD_자립)
		BREAK
	ENDIF
NEXT
IF YOME
	CALL DAIRY_EV22(YOME, CHILDNUM)
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV22(YOME, CHILDNUM)
#DIM YOME
#DIM CHILDNUM
#DIM 위자료
PRINTFORM %조사처리(CALLNAME:YOME,"로")%부터, 서당에서 %조사처리(CHILDNAME:YOME:CHILDNUM,"가")%

IF !RAND:4
	PRINTFORMW 다른 아이에게 부상을 입혔다고 전해들었다……
	PRINTFORML 설령 악의가 없었더라도, 보통사람 상대에게 %조사처리(CHILDNAME:YOME:CHILDNUM,"가")% 힘을 준다면 쉽게 상처를 줄 것이다
	PRINTFORML 부모님께 사과한 후에 %CHILDNAME:YOME:CHILDNUM%에게도 잘 말하게 해야……
	PRINTFORML %마스터는% 마음이 무거워졌다
	PRINTFORML
	CALL KOJO_MESSAGE_SEND("EVENT", 0, YOME, CHILDNUM, 0, "CHILD_DAIRY_서당", "다치게 했다")
	위자료 = MIN(10000, MONEY)
	MONEY -= 위자료
	CALL COLORMESSAGE(@"상대방에게 치료비로 {위자료} 지불했다", C_YELLOW, 1,1)
	CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 기력이 일시적으로 줄었다", C_YELLOW, 1,1)
	CALL BUFF_BASE(MASTER, BASE_기력,-1000)
ELSE
	PRINTFORMW 칭찬받았다고 전해들었다……
	PRINTFORML 과연 우리 아이다!
	PRINTFORML
	CALL KOJO_MESSAGE_SEND("EVENT", 0, YOME, CHILDNUM, 0, "CHILD_DAIRY_서당", "칭찬받았다")
	CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 기력이 일시적으로 늘었다", C_YELLOW, 1,1)
	CALL BUFF_BASE(MASTER, BASE_기력, 1000)
ENDIF
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV22 아이가_서당으로_t.ERB


code:
@DAIRY_CONDITION_EV24

SIF RAND:15 > 0
	RETURN -1
SIF DAY:3 >= 30
	RETURN -1

CALL DAIRY_EV24
RETURN 1

@DAIRY_EV24
PRINTFORMW 계절 이변이다！
FOR LOCAL,1,CHARANUM
	IF TALENT:LOCAL:요정
		CALL BUFF_BASE(LOCAL,BASE_체력,300)
		CALL BUFF_BASE(LOCAL,BASE_기력,300)
	ENDIF
NEXT
GatheringSeason = DAY:2 + 2
SIF GatheringSeason > 4
	GatheringSeason -= 4
PRINTFORMW 요정들의 체력과 기력이 올라가고、또한 %GET_MONTH(GatheringSeason)%의 소재를 추가로 채집할 수 있게 되었다！

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV24 계절이변.ERB


code:
@DAIRY_CONDITION_EV25

IF !FLAG:패션 && !RAND:20
	CALL DAIRY_EV25
	RETURN 1
ENDIF
RETURN -1

@DAIRY_EV25

#DIM Type, 2
#DIMS Desc, 2

Type:0 = 62
Desc:0 = 멋진 T셔츠
Type:1 = 74
Desc:1 = 세라복

LOCAL = RAND:2

PRINTFORML 아무래도 향간에는 %Desc:LOCAL%가 유행하고 있는 것 같다
FLAG:패션 = Type:LOCAL
;62 대신 원하는 의상세트 번호를 입력하면 모두의 복장을 바꿀 수 있다
;의상세트.ERB를 참조
PRINTW
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV25 패션_붐.ERB


code:
@DAIRY_CONDITION_EV26
#DIM 발생율
발생율 = 25
SIF FLAG:획득팬티총수 >= 4
	발생율 += SQRT(FLAG:획득팬티총수) / 2
발생율 += FLAG:독신 * 5
IF 발생율 < RAND:1000 || FLAG:이상기상
	RETURN -1
ENDIF
CALL DAIRY_EV26
RETURN 1

@DAIRY_EV26
FLAG:독신 = 0
PRINTFORMW %CALLNAME:MASTER%의 쌓아온 덕이 기적을 일으켰다……！
CALL COLORMESSAGE(@"분홍빛 바람이 불게 되었다",C_PINK,2,1)
FLAG:이상기상 = 8

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV26 분홍바람.ERB


code:
@DAIRY_CONDITION_EV2
IF !RAND:10 && TALENT:MASTER:120				;←여기에 발생 조건을 넣자 예로서 30분의 1으로 하고 있지만 적당하게 바꾸어
	CALL DAIRY_EV2
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV2
#DIMS MAS_HOME
PRINTFORML 한밤중, %마스터는% 행복한 감각에 눈을 떴다
PRINTFORMW 왠지 고간이 따뜻하다……
PRINTFORML !!!몽정해버리고 말았다!
PRINTFORMW 좀더 의미 있는 일을 할 걸 그랬다…　%마스터는% 깊게 후회했다
FLAG:71 = 0
TALENT:MASTER:120 = 0
BASE:MASTER:정력 -= 500
MAXBASE:MASTER:정력 +=1*3;;정력증가
EXP:MASTER:사정경험 ++
IF RAND:10 == 0
	PRINTFORML 기온, 풍향, 습도, 대기압, 요일, 월력, 행성의 나열 등 여러 조건들이 기적적으로 조합된 결과,
	FONTBOLD
	PRINTFORM %마스터가% 발한 정액은 결정화해,
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORMW 이 생성되었습니다!
	FONTREGULAR
	CALL GET_ITEM("미라다의돌", 1, 1)
	RETURN 1
ENDIF
CALL FISHER_YATES_SHAFFLE(CHARANUM)
LOCAL:1 = -1
MAS_HOME '= AREANAME_CHARA_HOME(MASTER)
FOR LOCAL, 0, CHARANUM
	;대상이 당신
	SIF SHAFFLE_ARRAY:LOCAL == MASTER
		CONTINUE
	;나오지 않는다
	SIF CFLAG:(SHAFFLE_ARRAY:LOCAL):출금
		CONTINUE
	;동거(同居) 하지 않아
	;판정이 다소 느슨하지만 전보다는 마시
	SIF AREANAME_CHARA_HOME(SHAFFLE_ARRAY:LOCAL) != MAS_HOME
		CONTINUE
	LOCAL:1 = SHAFFLE_ARRAY:LOCAL
	BREAK
NEXT
SIF LOCAL:1 <= 0
	RETURN

PRINTFORML %마스터가% 몽정으로 더럽힌 속옷을 빨고 있는데 %조사처리(CALLNAME:(LOCAL:1),"가")% 날아 왔다!
PRINTFORMW %조사처리(CALLNAME:(LOCAL:1),"는")% 그런 걸 다른 빨래와 함께 빨지 말라고 항의하고는 %CALLNAME:MASTER%의 손에 들려있는 속옷을 낚아채 돌아갔다…
JUEL:(LOCAL:1):11 += 1000
PRINTFORMW %CALLNAME:(LOCAL:1)%　욕정의 구슬＋1000
RETURN 1


code:
@DAIRY_CONDITION_EV3

IF FLAG:마지막에술집에방문한날 > 0 && !FLAG:술집매입강화 && !RAND:30
	CALL DAIRY_EV3
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV3
FLAG:술집매입강화 = 7
IF FLAG:신문구독
	PRINTFORML 붕붕마루 신문에 의하면, 마을 술집이 매입 강화 캠페인을 시작한 것 같다
	PRINTFORMW 평소보다 비싼게 사준다고 한다
ENDIF
RETURN 1


code:
@DAIRY_CONDITION_EV4

SIF RAND:7 > 0
	RETURN -1
CALL DAIRY_EV4
RETURN 1

@DAIRY_EV4
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	IF ABL:LOCAL:친밀 >= 5 && !CFLAG:LOCAL:신사거주
		PRINTFORMW 걷잡을 수 없는 사고에 빠져있던 %조사처리(CALLNAME:LOCAL,"는")%, 어느새 %마스터를% 생각하고 있었다는 것을 깨달았다
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 내일 %마스터를% 만나러 가기로 한 것 같다
		CFLAG:LOCAL:다음날내방 = 1
		BREAK
	ENDIF
NEXT


code:
@DAIRY_CONDITION_EV5

SIF RAND:(30 - TFLAG:OPPAI * 25) > 0
	RETURN -1

CALL DAIRY_EV5
RETURN 1

@DAIRY_EV5
#DIM 가슴신

;CASE 9는 미완성이니까
가슴신 = RAND:9
PRINTFORMW %마스터가% 꾸벅꾸벅 졸고 있자, 이상한 목소리가 들렸다
PRINTFORML
SETCOLOR C_AQUA
FONTBOLD
;미묘하게 변화시킨다
SELECTCASE 가슴신
	CASE 0 TO 4
		PRINTFORMW …들립니까… …들립니까… %CALLNAME:MASTER%… %CALLNAME:MASTER%…
		PRINTFORMW …지금…당신의 마음에…직접…말을 걸고 있습니다… 들립니까……
	CASE 5,6
		PRINTFORML …들립니까… …들립니까… %CALLNAME:MASTER%… %CALLNAME:MASTER%…
		PRINTFORMW …지금…당신의 마음에…직접…말을 걸고 있습니다… 들립니까……
	CASE 7,8
		PRINTFORML …들립니까……　들립니까… %CALLNAME:MASTER%…　%CALLNAME:MASTER%…
		PRINTFORMW …지금…당신의 마음에…직접…말을 걸고 있습니다… 들립니까……
	CASE 9
		PRINTFORML …들립니까… …들립니까… %CALLNAME:MASTER%… %CALLNAME:MASTER%…
		PRINTFORMW …지금……당신의 마음에…직접…말을 걸고 있습니다… 들립니까……
ENDSELECT
PRINTFORML
PRINTFORMW 당신은 큰 가슴과 작은 가슴, 어느 쪽을 좋아합니까……
RESETCOLOR
FONTREGULAR
;CALL COLORMESSAGE("당신은 큰 가슴과 작은 가슴, 어느 쪽을 좋아합니까……","AQUA", 1,1)
CALL ASK_M("큰 편", 1,"작은 편", 1,"가슴에 귀천이 없다", 1,"후타나리 자지를 좋아합니다", CFLAG:MASTER:누계질내정액 + CFLAG:MASTER:누계애널정액 + CFLAG:MASTER:누계정음 >= 5000)

SELECTCASE 가슴신
	CASE 0 TO 4
		IF RESULT == 3 && !가슴신
			CALL GOD_FUTANARI
		ELSE
			SETCOLOR C_AQUA
			FONTBOLD
			SIF RESULT == 3
				PRINTFORMW (무효 회답, 그리고)
			PRINTFORMW 앙케이트에 협조해 주셔서 감사합니다……사례비를 지불하겠습니다……
			RESETCOLOR
			FONTREGULAR
			PRINTFORML 캐바!
			PRINTFORMW %CALLNAME:MASTER%의 계좌에 \\1000 입금되었다!
			MONEY += 1000
		ENDIF
	CASE 5,6
	;거유의 신
		SELECTCASE RESULT
			CASE 0
				CALL OPPAIANSWER(1,1)
			CASE 1
				CALL OPPAIANSWER(1,-1)
			CASE 2
				CALL OPPAIANSWER(1,0)
			CASE 3
				CALL OPPAIANSWER(1,2)
		ENDSELECT
	CASE 7,8
	;작은 가슴의 신
		SELECTCASE RESULT
			CASE 0
				CALL OPPAIANSWER(-1,-1)
			CASE 1
				CALL OPPAIANSWER(-1, 1)
			CASE 2
				CALL OPPAIANSWER(-1, 0)
			CASE 3
				CALL OPPAIANSWER(-1, 2)
		ENDSELECT
	CASE 9
		SELECTCASE RESULT
			CASE 0,1
				CALL OPPAIANSWER(2,-1)
			CASE 2
				CALL OPPAIANSWER(2,2)
		ENDSELECT
ENDSELECT
PRINTFORMW

RETURN 1

@GOD_FUTANARI
CALL COLORMESSAGE("허허, 사람의 몸으로 여의 정체를 알아차리다니", C_D_RED, 2,1)
PRINTFORML
PRINTFORMW 목소리의 주인의 기색이 강대하게 바뀌었다……
CALL COLORMESSAGE("너의 후타나리 자지 사랑은 대단하구나", C_D_RED, 1,1)
CALL COLORMESSAGE("상으로, 한 가지 소원을 말해보거라", C_D_RED, 1,1)
$LOOP
CALL ASK_M("별로 괜찮습니다", 1,"후타나리 자지를 길러주세요", 1,"후타나리 자지를 강화해주세요", 1,"후타나리 자지를 빨게 해주세요", 1,"팬티 주세요", 1)
SELECTCASE RESULT
	CASE 0
		CALL COLORMESSAGE("멋없구나……자기가 어떤 뜻밖의 행운을 만났는지 모르는 모양이지", C_D_RED, 1,1)
		CALL COLORMESSAGE("뭐 괜찮다", C_D_RED, 1,1)
	CASE 1
		PRINTFORML 누구한테?
		DRAWLINE
		FOR LOCAL, 0, CHARANUM
			SIF TALENT:LOCAL:성별 == 1
				PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL, 15, LEFT%
		NEXT
		DRAWLINE
		PRINTFORML [999] 그만둔다
		INPUT
		IF INRANGE(RESULT, 0, CHARANUM) && TALENT:RESULT:성별 == 1
			CALL COLORMESSAGE("쉬운 일이지", C_D_RED, 1,1)
			PRINTFORMW %조사처리(CALLNAME:RESULT,"는")% 후타나리가 되었다
			TALENT:RESULT:성별 = 3
			MAXBASE:RESULT:정력 = 1000
		ELSE
			GOTO LOOP
		ENDIF
	CASE 2
		PRINTFORML 누구의?
		DRAWLINE
		FOR LOCAL, 1, CHARANUM
			SIF TALENT:LOCAL:성별 == 3 && MAXBASE:LOCAL:정력 < 1500
				PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL, 15, LEFT%
		NEXT
		PRINTL
		DRAWLINE
		PRINTFORML [999] 그만둔다
		INPUT
		IF INRANGE(RESULT, 1, CHARANUM) && TALENT:RESULT:성별 == 3 && MAXBASE:RESULT:정력 < 1500
			CALL COLORMESSAGE("쉬운 일이지", C_D_RED, 1,1)
			PRINTFORMW %CALLNAME:RESULT%의 정력 최대치가 증가했다
			MAXBASE:RESULT:정력 += 100
		ELSE
			GOTO LOOP
		ENDIF
	CASE 3
		CALL COLORMESSAGE("후후, 정말인지 훌륭한 후타나리 자지 사랑이여", C_D_RED, 1,1)
		CALL COLORMESSAGE("실컷 맛보도록, 받아들일 수 있다면", C_D_RED, 1,1)
		PRINTFORML 돌연 %CALLNAME:MASTER%의 입안에 거대한 질량이 발생하였다
		PRINTFORML 그것은 단단하고 부드러우며 짓눌릴 것 같은 강인함과 따뜻함을 느끼게 했다
		PRINTFORML 잠시 압도당했던 %마스터는% 정신을 차리고 인지를 초월한 것에 필사적으로 봉사한다
		PRINTFORML 무한하다고 생각되는 괴로움과 더 없는 행복의 시간 끝에 마침내 초상의 양물은 절정을 맞는다
		PRINTFORML 이루 말할 수 없을 만큼 진하고 좋은 맛과 방대한 신의 기운이 목구멍에 흩뿌려졌고 %마스터는% 의식을 잃었다……
		CALL BUFF_BASE(MASTER, BASE_기력, 2000)
		CALL BUFF_BASE(MASTER, BASE_체력, 2000)
	CASE 4
		CALL COLORMESSAGE("그런건 입지 않는다", C_D_RED, 2,1)
		GOTO LOOP
ENDSELECT
CALL COLORMESSAGE("잊지 마라, 여는 언제나 후타나리 자지와 함께 있다", C_D_RED, 1,1)
PRINTFORML 강대한 기색은 떠났다……

@OPPAIANSWER(ARG, ARG:1)
IF ARG:1 == -1
	;CALL COLORMESSAGE("멸망해나쁜 사람!","RED", 1,1)
	SETCOLOR C_RED
	FONTBOLD
	PRINTFORMW 망해라 나쁜 사람!!!
	RESETCOLOR
	FONTREGULAR
	PRINTFORML 엄청나게 성내는 소리와 함께 머리통을 두드려 깨트린 듯한 충격이 흘러 %CALLNAME:MASTER%의 의식은 끊어졌다
	BASE:MASTER:체력 -= 10000
ELSEIF ARG:1 == 2
	CALL COLORMESSAGE("뭐야 그거 무서워……", C_AQUA, 1,1)
	PRINTFORML 그 후 목소리는 끊어졌다
ELSEIF !ARG:1
		;이것이 정답의 분기를 다음에 만든다
		;PRINTFORML 당신은 마음 속 가슴을 사랑하고 있어요군요……
		SETCOLOR C_AQUA
		FONTBOLD
		PRINTFORMW 불쌍한 신념없는 사람……
		RESETCOLOR
		FONTREGULAR
			;CALL COLORMESSAGE("불쌍한 신념없는 사람……","AQUA", 1,1)
		PRINTFORML 그 후 목소리는 끊어졌다
ELSE
	SETCOLOR C_AQUA
	FONTBOLD
	PRINTFORML 그 마음을 잊지 않는한, \@ ARG < 0 ? 작은# 큰\@ 가슴은 언제나 당신의 편입니다…
	PRINTFORML 당신에게 \@ ARG < 0 ? 작은# 큰\@ 가슴의 가호가 있기를……
	RESETCOLOR
	FONTREGULAR
	CALL ASK_YN("가슴 만지게 해 주세요","축복해 주었으면 하는 가슴이 있다")
	IF RESULT
		PRINTFORML 누구의?
		DRAWLINE
		FOR LOCAL, 0, CHARANUM
			IF ARG > 0
				SIF TALENT:LOCAL:가슴사이즈 < 2
					PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL%
			ELSE
				SIF TALENT:LOCAL:가슴사이즈 > -2
					PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL%
			ENDIF
		NEXT
		PRINTFORML
		DRAWLINE
		PRINTFORML [999] 그만둔다
		INPUT
		SELECTCASE RESULT
			CASE 0 TO CHARANUM
				SETCOLOR C_AQUA
				FONTBOLD
				PRINTFORMW %CALLNAME:RESULT%의 가슴에 축복을……
				RESETCOLOR
				FONTREGULAR
				IF ARG > 0
					TALENT:RESULT:가슴사이즈 ++
					PRINTFORMW %CALLNAME:RESULT%의 가슴이 커졌다!
				ELSE
					TALENT:RESULT:가슴사이즈 --
					PRINTFORMW %CALLNAME:RESULT%의 가슴이 작아졌다!
				ENDIF
			CASEELSE
		ENDSELECT
	ELSE
		;비비게 해 줄 수 있는 분기는 미완성
		SETCOLOR C_AQUA
		FONTBOLD
		PRINTFORML 당신의 가슴 사랑이 더 강해질때 주무르게 해드리죠……
		IF !TALENT:MASTER:능숙한손가락
			PRINTFORMW 그 대신, 가슴을 더 깊게 사랑하는 법을 전수합니다……
			RESETCOLOR
			FONTREGULAR
			PRINTFORMW %마스터는%【능숙한손가락】을 얻었다!
			TALENT:MASTER:능숙한손가락 = 1
		ENDIF
		RESETCOLOR
		FONTREGULAR
	ENDIF
	PRINTFORML 가슴에 싸일 것 같은 기분 속에서 %CALLNAME:MASTER%의 의식은 꿈속에 빠져들었다……
ENDIF

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV5 OPPAI_t.ERB


code:
@DAIRY_CONDITION_EV6

IF !RAND:10
	CALL DAIRY_EV6
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV6
;여기에 이벤트 내용을 쓰자 최후는 RETURN 1해
LOCAL = RAND:5
SELECTCASE LOCAL
	CASE 0
		LOCALS = 청소
		FLAG:가호 = 70
	CASE 1
		LOCALS = 수련
		FLAG:가호 = 74
	CASE 2
		LOCALS = 학문
		FLAG:가호 = 75
	CASE 3
		LOCALS = 요리
		FLAG:가호 = 76
	CASE 4
		LOCALS = 연주
		FLAG:가호 = 79
ENDSELECT
PRINTFORMW 꿈속에서 %마스터는% 익숙한 신의 목소리를 들었다
PRINTFORML
SETCOLOR C_YELLOW
FONTBOLD
PRINTFORMW ((나를 믿는 자여……))
PRINTFORMW ((내일은 %LOCALS%에 힘써라……　%LOCALS%에 힘쓰는 것이다……))
RESETCOLOR
FONTREGULAR
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV6 가호_t.ERB


code:
@DAIRY_CONDITION_EV7

IF !RAND:15 && MAIN_MAP == 9
	CALL DAIRY_EV7
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV7
#DIM 인원수
#DIM YUGI
VARSET LOCAL
YUGI = !RAND:4
IF YUGI
	LOCALS = 유기와 오니들
ELSE
	LOCALS = 오니들
ENDIF
인원수 = 1 + RAND:5
PRINTFORML %마스터가% 잠자리에 들려고 했을 때, 술을 마신 {인원수}명의 오니가 기어 들어왔다!
SIF YUGI
	PRINTFORML 게다가 유기까지 있다……
PRINTFORMW 아무래도 여기 눌러앉아서 계속 마시려고 하는 것 같다……
PRINTFORML
;다른 선택지는 다음에
;ASK_M("받아들이는","술을 건네주어 나가있어 받는","")
PRINTFORML %마스터는% 자포자기로 %조사처리(LOCALS,"를")% 받아들였다
PRINTFORML %조사처리(LOCALS,"는")% 조금도 사양하지 않고 가지고 온 술과 안주를 펼쳐놓고 야단법석을 떨기 시작했다
LOCAL:1 = 1
FOR LOCAL, 0, 인원수
	LOCAL:1 += RAND:3
NEXT
SIF YUGI
	LOCAL:1 += RAND:3 + RAND:3

FOR LOCAL, 0, LOCAL:1
	PRINTFORMW …
	PRINTFORMW ……
	PRINTFORML %마스터는% 억지로 술을 먹게 되었다
	SELECTCASE RAND:3
		CASE 0
			PRINTFORMW 평범하게 좋은 술이다
			LOCAL:2 = 200
		CASE 1
			PRINTFORMW 꽤 독한 술이다
			LOCAL:2 = 400
		CASE 2
			PRINTFORMW 이건 독해!
			LOCAL:2 = 700
	ENDSELECT
	PRINTFORMW
	BASE:MASTER:술기운 += LOCAL:2
	IF YUGI
		CALL CHANGE_CFLAG(2,64,3)
		CALL CHANGE_CFLAG(4,64,1)
	ENDIF
	IF BASE:MASTER:술기운 > MAXBASE:MASTER:술기운
		PRINTFORML 「우웩!」
		PRINTFORML %마스터는% 토했다!
		YOGORE:(CFLAG:MASTER:초기위치) += 1500
		BASE:MASTER:기력 -= 500
		MAXBASE:MASTER:술기운 += 3
		PRINTFORMW %조사처리(LOCALS,"는")% 껄껄 웃고 있다…
		IF BASE:MASTER:기력 < 0
			PRINTFORMW %마스터는% 쓰러졌다!
			BASE:MASTER:기력 -= 1000
			LOCAL:3 ++
			BREAK
		ELSE
			BASE:MASTER:술기운 /= 2
		ENDIF
	ELSEIF BASE:MASTER:술기운 > MAXBASE:MASTER:술기운 * 3 / 4
		PRINTFORMW 기분 나빠졌다…
	ENDIF
NEXT

IF !LOCAL:3
	PRINTFORML %마스터는% 집요한 술 강요를 견뎌냈다!
	IF YUGI
		CFLAG:64:호감도 += 150
		CFLAG:64:신뢰도 += 100
	ENDIF
	MAXBASE:MASTER:술기운 += 20
ENDIF
CFLAG:MASTER:철야 = 1
YOGORE:(CFLAG:MASTER:초기위치) += 1000
MAXBASE:MASTER:술기운 = LIMIT(MAXBASE:MASTER:술기운, 1500,3000)
PRINTFORML 새벽녘 %조사처리(LOCALS,"는")% 떠났다
PRINTFORML 이후 잠이 부족한 %마스터와% 대량의 빈병\@ LOCAL:3? 과 토사물의 냄새가# 이\@ 남겨졌다
PRINTW
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV7 음주강요_t.ERB


code:
@DAIRY_CONDITION_EV8

IF FLAG:밭일성과 > 60 && TALENT:MASTER:2 & 2 && FLAG:야채낭데일리 != 2 && !RAND:20
	CALL DAIRY_EV8
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV8

PRINTFORML %마스터가% 잠자리에 들려고 할 때, 낯선 소녀가 찾아왔다
PRINTFORML %CALLNAME:MASTER%에게 항상 신세를 지고 있어 감사의 뜻을 표하고 싶다고 한다
IF !FLAG:야채낭데일리
	PRINTFORML 그러나 %마스터는% 전혀 짐작이 가지 않는다…
ELSE
	PRINTFORML 전에도 비슷한 일이 있었군…
ENDIF
CALL ASK_YN("사람 잘못봤다","차려진 밥상을 먹지 않을리가")
IF !RESULT
	IF !FLAG:야채낭데일리
		PRINTFORML 미인계인가 뭐냐, 잘못하면 요괴의 덫일지도 모른다
		PRINTFORML 그렇게 생각한 %마스터는% 소녀를 되돌려 보냈다
		PRINTFORML 소녀는 유감스러운 듯 떠났다
	ELSE
		PRINTFORML 왠지 무서워서 그만 두기로 했다
		PRINTFORML 소녀는 유감스러운 듯 떠났다
	ENDIF
ELSEIF BASE:MASTER:기력 < 500 || BASE:MASTER:체력 < 750 || BASE:MASTER:정력 < 300
	PRINTFORML 그러나 졸려서 물러갔으면 했다
	PRINTFORML 소녀는 유감스러운 듯 떠났다
ELSE
	IF !FLAG:야채낭데일리
		PRINTFORML %마스터는% 소녀를 찬찬히 보았다
		PRINTFORML 약간 통통한 계열로 그럭저럭 사랑스럽다
		PRINTFORML 이런 시간에 남자가 있는 곳에 오다니 감사의 뜻이란 즉 그런 말일 것이다, 사정은 모르겠지만 일단 하고나서 생각할까
	ELSE
		PRINTFORM 할 수 있는데다가 작물이 자란다면 이렇게 손해보지 않는 이야기는 없다,
	ENDIF
	PRINTFORML 그렇게 생각한 %마스터는% 소녀를 방으로 불러들였다
	PRINTFORMW 그리고 답례로 무엇을 해줄 수 있는지 묻자 기대대로 소녀는 자기 옷에 손을 대었다…
	PRINTFORMW …
	PRINTFORMW ……
	IF !FLAG:야채낭데일리
		PRINTFORML 새벽녘, 그녀는 비몽사몽한 %마스터를% 두고 떠났다
		PRINTFORML 결국 신세를 지고 있다니 무슨 일이었을까?
		PRINTFORMW 떠나면서 「아이를 많이 낳는다」고 불온한 말을 했던 것 같은데……
		PRINTFORMW …
		PRINTFORMW ……
		PRINTFORMW 밭을 돌보러 나온 %마스터는% 위화감을 깨달았다
		PRINTFORMW 갑자기 열매가 좋아진 것 같다……
		PRINTFORML ……이거, 먹어도 괜찮을까?
	ELSE
		PRINTFORML 예상과 다름없이, 다음날 아침 본 밭은 더 나은 생육을 보이고 있었다
		PRINTFORML 어제 내놓은 정액이 이 열매가 되어 수확되고 출하되어 가는 것인가……
		PRINTFORML %마스터는% 묘한 감회에 빠졌다
	ENDIF
	BASE:MASTER:정력 -= 300
	BASE:MASTER:체력 -= 200
	FLAG:밭일성과 += 20
	FLAG:야채낭데일리 = 2
ENDIF
PRINTW
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV8 야채낭_t.ERB


code:
@DAIRY_CONDITION_EV9

IF !RAND:20 && MAIN_MAP == 9
	CALL DAIRY_EV9
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAIRY_EV9
PRINTFORML 지령전의 개가 %CALLNAME:MASTER%의 이불 속으로 잠입해 왔다
;ある人のところ :ある人がいる場所。(住んでいる場所や、いつも仕事をしている場所など）
;어떤 사람이 있는 곳.(사는 장소나, 항상 일을 하는 장소 등)
;사토리가 있는 곳 = 지령전
SELECTCASE DAY:2
	CASE 2
		PRINTFORML 잠들기 힘들다……
		CFLAG:MASTER:철야 = 1
	CASE 4
		PRINTFORML 따뜻하다……
		CALL RECOVER(MASTER, 300,"체력")
		CALL RECOVER(MASTER, 500,"기력")
	CASEELSE
		CALL RECOVER(MASTER, 100,"체력")
		CALL RECOVER(MASTER, 200,"기력")
ENDSELECT
PRINTW
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV9 개동침_t.ERB


code:
@만남판정
#DIM 기본만남율
#DIM 만남율

만남율 = 0
FOR LOCAL, 1, CHARANUM
	;자택 거주 캐릭터라면 조우하지 않는다
	SIF CFLAG:LOCAL:신사거주 > 0
		CONTINUE
;방문자 각각의 만남 플래그 초기화
	CFLAG:LOCAL:만난위치 = 0
NEXT
LOCAL:2 = CHARANUM

;자주(잘) 생각하면(자) 이 계산 몇 번도 하지 않아 좋은 일을 눈치챘다
;캐릭터수 증가할 때마다 조정 하는 것도 뭐하고,%로 설정하는 것이 좋으면 좋은 생각이 들었으므로 변경
;%과 만남 확률의 밸런스 조정은 내던지는 것(작자님에게 맡깁니다)
SELECTCASE (FLAG:출금인원수 * 100 / CHARANUM)
	CASE IS >= 60
		기본만남율 = 26
	CASE IS >= 45
		기본만남율 = 28
	CASE IS >= 30
		기본만남율 = 30
	CASE IS >= 15
		기본만남율 = 32
	CASE IS >= 10
		기본만남율 = 34
	CASEELSE
		기본만남율 = 36
ENDSELECT

FOR LOCAL, 1, CHARANUM
	;자택 거주 캐릭터라면 조우하지 않는다
	SIF CFLAG:LOCAL:신사거주 > 0
		CONTINUE
	IF !(CFLAG:LOCAL:거점내방 && CFLAG:LOCAL:연회참가)
		;10이하가 되면 Default 설정이라고 확정으로 만남 플래그가 설 뿐(만큼)이므로(0 이하가 아니면) 문제 없음
		만남율 = 기본만남율 - (TALENT:LOCAL:사모 + TALENT:LOCAL:연모 * 2)
		SIF CFLAG:LOCAL:의뢰상황
			만남율 -= 3
		SELECTCASE LOCAL
			;캐릭터별로 만남율을 바꾸는 경우는 CASE를 더하는·Default는10/(기본만남율)
			CASEELSE
				SIF RAND:만남율 + 1 <= 10
					CALL 만남플래그결정(LOCAL)
		ENDSELECT
	ENDIF
	SIF CFLAG:LOCAL:현재위치 == SUKIMA() && CFLAG:LOCAL:신사거주
		CFLAG:LOCAL:현재위치 = CFLAG:LOCAL:신사거주
NEXT

@만남플래그결정(ARG)
;CFLAG:ARG:만난위치 = ARG의 만남 지점
;LOCAL = 산출된 가지점(만남 지점/10)
;LOCAL:1 = 랜덤에 결정할 수 있던 지역

;25%로 자택에 출현. 자택이 미해금의 경우는 스킵
;이 후 자택이 될 가능성도 있으므로 실제로는 좀 더 높다
IF !RAND:4 && INRANGE(CFLAG:ARG:자택위치 , 100 , 1100)
	CFLAG:ARG:만난위치 = CFLAG:ARG:자택위치
;33%(자택이 있는 경우 약25%)로 자주가는지역. 자주가는지역이 미설정의 경우는 스킵
;이 후 자주가는지역이 될 가능성도 있으므로 실제로는 좀 더 높다
ELSEIF !RAND:2 && CFLAG:ARG:자주가는지역 >= 0 && CFLAG:ARG:자주가는지역 != MAIN_MAP
	CALL FISHER_YATES_SHAFFLE(9)
	;1~10의 숫자를 셔플 해 차례로 당겨 체크, 최대 10회의 판정으로 끝날 것.
	FOR LOCAL:10 , 0 , 9
		LOCAL = CFLAG:ARG:자주가는지역 * 10 + SHAFFLE_ARRAY:(LOCAL:10) + 1
		;출현 장소나 체크
		CALLFORM CHARAMOVE_DATA_{ARG}(7, LOCAL)
		;출현 장소가 아니면 재추첨(출현 장소가 설정되어 있지 않으면 만난위치가 변경되지 않기 때문에 주의)
		SIF !RESULT
			CONTINUE
		;출현 장소로 설정
		BREAK
	NEXT
	;장소를 설정
	;만일 자주가는지역내에 출현 장소가 설정되어 있지 않았던 경우, 마지막에 당긴 장소가 들어가도록(듯이)
	CFLAG:ARG:만난위치 = LOCAL * 10
;자주가는지역이 있는 경우66%·자택도 있는 경우 약50%·자택만 있는 경우75%·아무것도 없는 경우100%
;자택·자주가는지역도 포함해 선출
ELSE
	CALL FISHER_YATES_SHAFFLE(11)
	FOR LOCAL:10 , 0 , 11
		;랜덤에 지역 선택
		LOCAL:1 = SHAFFLE_ARRAY:(LOCAL:10)
		;거점 맵이라면 재추첨
		SIF LOCAL:1 == MAIN_MAP
			CONTINUE
		;CHARAMOVE를 참조해 가지 않는 지역이라면 재추첨
		RESULT = 0
		CALLFORM CHARAMOVE_DATA_{ARG}(6, LOCAL:1)
		SIF !RESULT
			CONTINUE
		BREAK
	NEXT
	;가는 지역이 설정되어 있지 않았던 경우, 마지막에 당긴 지역에 간다

	CALL FISHER_YATES_SHAFFLE(9)
	FOR LOCAL:10 , 0 , 9
		;랜덤에 장소를 선택
		LOCAL = LOCAL:1 * 10 + SHAFFLE_ARRAY:(LOCAL:10) + 1
		;CHARAMOVE를 참조해 가지 않는 지역이라면 재추첨
		RESULT = 0
		CALLFORM CHARAMOVE_DATA_{ARG}(7, LOCAL)
		SIF !RESULT
			CONTINUE
		SIF STR:(6000 + LOCAL) == ""
			CONTINUE
		BREAK
	NEXT
	;출현 장소로 설정
	;만일 출현 장소가 설정되어 있지 않았던 경우, 마지막에 당긴 장소가 들어가도록(듯이)
	CFLAG:ARG:만난위치 = LOCAL * 10
ENDIF

@출현상태(ARG)
;=1거점에는 오지 않는 =2오는 =3 살고 있다
#FUNCTION
IF CFLAG:ARG:출금 || TALENT:ARG:스침
	RETURNF 0
ELSEIF CFLAG:ARG:신사거주
	RETURNF 3
ELSEIF CFLAG:ARG:거점내방
	RETURNF 2
ELSEIF CFLAG:ARG:연회참가
	RETURNF 2
ELSEIF !CHARA_HOLIDAY(ARG)
	RETURNF 1
ELSEIF CFLAG:ARG:초대
	RETURNF 1
ELSEIF CFLAG:ARG:만난위치
	RETURNF 1
ELSE
	RETURNF 0
ENDIF


code:
;모리야 신사 제작 이치방쿠지, 마침내 판매 개시입니다!
;
;룰
;1. 복권은 전부 300장, 다 뽑거나 특상을 뽑으면 상자가 교환된다
;2. 최초로 뽑을 때, 혹은 상자의 교체시에 어느 차례로 무엇이 나올까는 확정하고 있다(셔플 배열을 전용의 복권 배열에 Copy(복사))
;3. 복권의 결과를 확인했더니 당긴 복권을 폐기해 전 채우고 한다
;4. 특상은 1개, 뽑으면 SETBIT FLAG:999, 3 으로 상자 교환 플래그를 세운다
;특상의 약이 어느 쪽인지는 FLAG:999의 제4 비트로 판정(판정 후 INVERTBIT로 플래그를 바꿔 넣는다)
;유저 변수를 세이브할 수 없기 때문에 SAVESTR를 1개 빌려 거기에 결과를 캐릭터라인으로 남겨 둔다
;(16/07/02) 사양 변경에 의해 유저 변수의 세이브가 가능하게 되었으므로 SAVESTR의 이용을 정지
@FOTUNE_MORIYA
#DIM RANK_FORTUNE
#DIM 상대
#DIM SKIP_FLAG
#DIMS 복권매수, 300

IF MATCH(MORIYA_BOX, 0) >= 2 && STRLENS(SAVESTR:9)
	SPLIT SAVESTR:9, ",", 복권매수
	FOR LOCAL:0, 0, 300
		MORIYA_BOX:(LOCAL:0) = TOINT(복권매수:(LOCAL:0))
	NEXT
	SAVESTR:9 = ""
ENDIF

;토부「개운대문」
WHILE TFLAG:194 && MORIYA_BOX:0 >= 60
	SETCOLOR COLOR ("YELLOW")
	PRINTFORMW 토부 「개운대문」이 발동!
	RESETCOLOR
	ARRAYSHIFT MORIYA_BOX, -1, -1
	TFLAG:194 --
WEND

;하쿠레이 운세뽑기에 의한 행운보정, 결과가 나쁘면 그것을 소멸해 다시 당긴다
DO
	SKIP_FLAG = 0
	SELECTCASE TFLAG:행운보정
		CASE IS <= 0
		;수세미 회피
		CASE 1 TO 2
			SIF MORIYA_BOX:0 >= 130
					SKIP_FLAG = 1
		;7등까지를 회피
		CASE 3 TO 4
			SIF MORIYA_BOX:0 >= 60
					SKIP_FLAG = 1
		;5등까지를 회피
		CASE 5
			SIF MORIYA_BOX:0 >= 25
					SKIP_FLAG = 1
		CASEELSE
	ENDSELECT
	IF SKIP_FLAG
		ARRAYSHIFT MORIYA_BOX, -1, -1
		PRINTFORML 하쿠레이 신사에서 좋은 운세를 뽑았으니, 아마 좋은 게 당첨될 것이다
		TFLAG:행운보정 --
	ENDIF

LOOP SKIP_FLAG

;복권의 결과는 0~299. 이것을 각 상에 배분해 둔다
;내역 특상:1책, 1등:3개, 2등:5개, 3등:6개, 4등:10개, 5등:15개, 6등:20개, 7등:30개, 8등:40개, 9등 (수세미):70개, 10등 (스카):100개
;RANK_FORTUNE를 MORIYA_BOX:0에 의존시키지 않게(하려고 생각하면 기존의 것을 만지작거리지 않고 아래에 추가할 수 있어. RANK_FORTUNE가 혼란하기 때문에 하지 않는다고 생각하는데)
SELECTCASE MORIYA_BOX:0
;특상：홀수 번째 라면 체력, 짝수회 라면 기력이 오르는 환약(최대치＋50)
CASE 0
	SETBIT FLAG:999, 3
	RANK_FORTUNE = 0
;1등：모리야신사의 2주＋카제하후리과의 데이트 티켓. 각 1매
CASE 1 TO 3
	상대 = MORIYA_BOX:0 + 30
	RANK_FORTUNE = 1
;2등：코치야 사나에님이 밤행복의 기적(전원 좋은 기분＋자신의 신분 회복)
CASE 4 TO 8
	RANK_FORTUNE = 2
;3등：MORIYA 부트 캠프 초대권, 불과 1시간에 몰라 보는 것 같은 효과가!
CASE 9 TO 14
	RANK_FORTUNE = 3
;4등：즐겁게 감추자, 환상향 감추어 비교 세트
CASE 15 TO 24
	RANK_FORTUNE = 4
;5등：건강에 흥미없는 당신에게, 건강 발랄∼？세트
CASE 25 TO 39
	RANK_FORTUNE = 5
;6등：이것으로 그 아이와 이이사이에!사랑의 해프닝 세트
CASE 40 TO 59
	RANK_FORTUNE = 6
;7등：소득의 기쁨을 당신에게, 야채의모종 세트(야채가 생긴다고는 말하지 않았다)
CASE 60 TO 89
	RANK_FORTUNE = 7
;8등：한 번 먹으면 멈추지 않는, 의리 초콜렛의 임금님의 엔트리다!
CASE 90 TO 129
	RANK_FORTUNE = 8
;유감…9등：수세미, 10개 모으면 돈의 수세미와 교환입니다!
CASE 130 TO 199
	RANK_FORTUNE = 9
;스카：꽝 복권, 5매 모으면 다시 한 번 파합니다
CASE 200 TO 299
	RANK_FORTUNE = 10
CASEELSE
	RANK_FORTUNE = 99
ENDSELECT
;뽑은 복권은 폐기처분해 전 채워 빈 곳은―1으로 채운다
ARRAYSHIFT MORIYA_BOX, -1, -1
;SAVESTR:9 = {MORIYA_BOX:0}
;FOR LOCAL:0, 1, 300
;	SAVESTR:9 = %SAVESTR:9%,{MORIYA_BOX:(LOCAL:0)}
;NEXT
PRINTDATAW
	DATAFORM 「자, 결과는…」
	DATAFORM 「으음, 이건……」
	DATAFORM 「아―… 아―？ 아―, 아―!」
	DATAFORM 「어디 보자, 이건…」
	DATAFORM 「이게 결과인가요……」
	DATAFORM 「뭐가 나올까, 뭐가 나올까?」
	DATAFORM 「어라라, 이걸 뽑았나요」
	DATAFORM 「어디 보자, 이건 말이죠…」
	DATAFORM 「이상한 글씨가…, 아래위 뒤집어 봤군요」
	DATAFORM 「호오호오, 이건 제법 잘…」
	DATAFORM 「이런 게 나왔는데요∼」
	DATAFORM 「……? ……! ………!」
ENDDATA
PRINTW
;결과 처리. 지문까지 기입하고 있습니다
SELECTCASE RANK_FORTUNE
;특상：홀수 번째 라면 체력, 짝수회 라면 기력이 오르는 환약(최대치＋50)
CASE 0
	PRINTW 마땅찮은 표정을 짓다가 갑자기 함박웃음을 머금는다
	PRINTL 「축하합니다―! 대박 당첨됐습니다!」
	PRINTFORML 「특상이신 손님께는 놀랍게도 \@GETBIT(FLAG:999, 4) ? 활력의 환약 # 생명의 비분 \@을 드립니다」
	PRINTL 「자, 꿀꺽 드셔보세요. 아, 물은 여기 있어요」
	PRINTW 건네받은 약을 물과 함께 마시자 몸속에서부터 힘이 솟구치는 느낌이 든다……
	IF GETBIT(FLAG:999, 4)
		MAXBASE:MASTER:기력 += 50
	ELSE
		MAXBASE:MASTER:체력 += 50
	ENDIF
	PRINTFORML 힘이 솟구치던 감각이 사라지자 당신은 \@GETBIT(FLAG:999, 4) ? 기력 # 체력 \@의 한계를 살짝 넘은 것 같았다
	INVERTBIT FLAG:999, 4

	PRINTFORMW 「맞아맞아, 깜빡 잊고 말 안 했네요」
	PRINTFORML 「특상이 이 상자에서 나왔으니 희망하신다면 다음 복권 상자를 준비하겠습니다」
	PRINTFORML 「새 상자로 바꾸고 싶어지면 뽑기 전에 말해 주세요」
	PRINTFORMW 「단, 새 상자로 바꾸면 현재 상자는 폐기하니 주의해주세요」
;1등：모리야신사의 2주＋카제하후리과의 데이트 티켓. 각 1매
CASE 1
	PRINTW 「이것은……」
	IF CFLAG:상대:출금
		PRINTFORML 「죄송합니다. 이건 1등의 %NAME:상대% 님과의 데이트권입니다만, %NAME:상대%님은 현재 아무래도 형편이 닿지 않아 이 권리를 이행하실 수가 없습니다」
		PRINTFORML 「대신할 상품을 이하 중에서 선택해주세요」
		CALL ASK_M("환불", 1,@"%NAME:31%과의 데이트권",!CFLAG:31:출금,@"%NAME:32%과의 데이트권",!CFLAG:32:출금,@"%NAME:33%과의 데이트권",!CFLAG:33:출금,"필요 없어", 1)
		SELECTCASE RESULT
			CASE 0
				PRINTFORML 「그러면 환불로 처리하겠습니다」
				PRINTFORML 「이번엔 저희 실수로 폐를 끼쳤습니다」
				MONEY += 3000
				RETURN 0
			CASE 1 TO 3
				상대 = RESULT + 30
				GOTO DATE_RIGHT
			CASE 4
				PRINTFORML 「그럴 수가, 그럼 제 기분이 안 풀려요……!」
				PRINTFORML 「최소한 이거라도 이것을 받아주세요」
				PRINTFORML 꽝 복권 다발을 떠넘겼다……
				ITEM:모리야부적 += 50
				RETURN 0
		ENDSELECT
	ENDIF
	PRINTL 쓴웃음 지으면서 안쪽에 얘기한다
	$DATE_RIGHT
	IF CFLAG:상대:수면 || CFLAG:상대:쇠약
		PRINTFORML 「네? 졸려? 지쳤어? 거기를 어떻게든」
		PRINTFORML 「조금 기다려 주세요. 금방 돌아올 테니까요」
		PRINTFORMW 종종걸음으로 안쪽에 달려 갔다
		PRINTFORMW ……
		PRINTFORMW 불명료하면서, 그러고도 신이신가요, 좋아해 일하고 있는 것이 아닌 것은 누구라도 같은, 여기가 신앙의 돈벌이 어디, 라고 하는 말이 들린다……
		PRINTFORMW ……
		PRINTFORMW 돌아왔다
		PRINTFORMW 「후우, 기다리게 해서 죄송합니다」
		CFLAG:상대:수면 = 0
		CFLAG:상대:쇠약 = 0
		BASE:상대:체력 = MAXBASE:상대:체력 / 2
		TALENT:상대:기분 = -1
	ENDIF
	PRINTFORML 「축하합니다. 1등의 %NAME:상대%님과의 데이트권, 당선입니다!」
	PRINTW 「덧붙여 이 권리는 즉시 행사되기 때문에 즐겨 와 주세요♪」
	IF (FLAG:데이트상대 && GET_MAPID(CFLAG:MASTER:현재위치) != MAIN_MAP)
		CFLAG:(FLAG:데이트상대):데이트중 = MAIN_MAP
		CFLAG:(FLAG:데이트상대):동행 = 0
		;연속으로 적용한 호운의 군사를 위해서(때문에)…
		CALL 모리야복권데이트상대의리셋()
	ELSEIF GET_MAPID(CFLAG:MASTER:현재위치) == MAIN_MAP
		FOR LOCAL, 1, CHARANUM
			SIF !CFLAG:LOCAL:동행
				CONTINUE
			CFLAG:LOCAL:동행 = 0
			SIF MATCH(LOCAL, 31,32,33)
				CALL 모리야복권데이트상대의리셋()
		NEXT
	ENDIF
	CFLAG:상대:현재위치 = CFLAG:MASTER:현재위치
	;상대의 캐릭터의 동행(MASTER도), 현재위치, 데이트중의 CFLAG를 세트
	IF GET_MAPID(CFLAG:MASTER:현재위치) != MAIN_MAP
		CFLAG:MASTER:동행 = 60
		CFLAG:상대:동행 = 60
		CFLAG:상대:데이트중 = CFLAG:MASTER:현재위치 / 100
		FLAG:데이트상대 = 상대
		;모리야복권데이트상대에 데이트의 상대를 격납
		FLAG:모리야복권데이트상대 = FLAG:데이트상대
		CFLAG:(FLAG:데이트상대):내방시간 = 0
		CFLAG:(FLAG:데이트상대):귀가시간 = 0
		CFLAG:(FLAG:데이트상대):취침시간 = 0
		CFLAG:(FLAG:데이트상대):기상시간 = 0
	;메인 맵시는 최대 3시간까지
	ELSE
		CFLAG:MASTER:동행 = 180
		CFLAG:상대:동행 = 180
		FLAG:모리야복권데이트상대 = 상대
	ENDIF
	SIF !CFLAG:(상대):안면
		CALL KOJO_MESSAGE_SEND("ENCOUNTER", MASTER, FLAG:모리야복권데이트상대, 1,0)
;2등：행복의 기적이 환상향을 쌉니다(전원 좋은 기분＋자신의 신분 회복)
CASE 2
	PRINTL 「축하합니다. 2등은 코치야 사나에 님께서 베푸는 행복의 기적의 실연입니다」
	PRINTW 「언제나 츤츤대는 그 소녀도 오늘만은 데레해줄지도?」
	FOR LOCAL:0, 1, CHARANUM
		TALENT:(LOCAL:0):기분 = 1
		BASE:(LOCAL:0):분노 = 0
	NEXT
	PRINTL 마음이 매우 온화하게 된 것 같았다
	PRINTL 「아, 데레해준다고 해서 도를 넘은 짓을 하면 화내니 조심하세요」
	PRINTL 기적도 그렇게까지 만능은 아닌 모양이다
	PRINTW 덩달아 자신의 몸이 가벼워진 것 같다
	BASE:MASTER:체력 = MAXBASE:MASTER:체력
	BASE:MASTER:기력 = MAXBASE:MASTER:기력
	BASE:MASTER:정력 = MAXBASE:MASTER:정력
	BASE:MASTER:TSP  = MAXBASE:MASTER:TSP
	BASE:MASTER:술기운 = 0

;3등：MORIYA 부트 캠프 초대권, 불과 1시간에 몰라 보는 것 같은 효과가!
CASE 3
	PRINTL 「3등 당선 축하합니다?」
	PRINTL 「3등은 MORIYA 부트 캠프 우대권이 됩니다」
	PRINTW 「또한 사용 기한은 금일 내이므로 지금부터 개시하겠습니다」
	PRINTL 아니, 지금부터 다른 예정이……
	PRINTW 「거부권은 없어요. 자자, 이리로……」
	PRINTL ………
	PRINTL ……
	PRINTL …
	;회화경험, 전투경험, 학습경험의 나 가장 낮은 것을 끌어 올린다
	;또한 강사는 회화/전투/학습으로 사나에/카나코/스와코가 담당
	;……케로 쪽이 전투 높고 카나코와 교양이 같은의는 비밀이야
	;아, 내친 김에 강사에도 경험을 추가. 가르치는 것도 좋은 경험이고
	IF EXP:MASTER:회화경험 < EXP:MASTER:전투경험 &&  EXP:MASTER:회화경험 < EXP:MASTER:학습경험
		PRINTW 사나에에 의한 말투 강좌에 의해 회화의 패턴을 잡았다!
		EXP:MASTER:회화경험 += 10
		EXP:31:회화경험 += 5
	ELSEIF EXP:MASTER:전투경험 < EXP:MASTER:학습경험
		PRINTL 카나코 직접의 지도에 의해 전투 기술을 두드려 난처할 수 있었다!
		PRINTW 내친 김에 훈련의 피로도 낫자마자 해 주었다
		MAXBASE:MASTER:체력+= 10
		EXP:MASTER:전투경험 += 10
		EXP:32:전투경험 += 5
	ELSE
		PRINTW 스와코의 정중한 담아 교육은 반드시 향후 도움이 될 것이다
		MAXBASE:MASTER:기력+=10
		EXP:MASTER:학습경험 += 10
		EXP:33:학습경험 += 5
	ENDIF
	;그토록 가혹한 훈련에서도 불과 1시간, MORIYA 부트 캠프, 오니(두려운)……
	TIME += 60

;4등：즐겁게 감추자, 환상향 감추어 비교 세트
CASE 4
	PRINTL 「4등이 맞았습니다」
	PRINTL 「4등의 경품은 환상향 감추어 비교 세트입니다」
	PRINTL 「카나코 님 비장의 감주에 탁주와 청주, 그리고 오니고로시를 준비했습니다」
	PRINTW 「해 밤 비교해!」
	ITEM:감주 = MIN(ITEM:감주 + 1, 99)
	ITEM:탁주 = MIN(ITEM:탁주 + 1, 99)
	ITEM:청주 = MIN(ITEM:청주 + 1, 99)
	ITEM:오니고로시 = MIN(ITEM:오니고로시 + 1, 99)
	PRINTW 「아, 가지지 못했던 만큼은 카나코 님에게 돌려주지 않으면 안 되기 때문에 두고 가 주세요」

;5등：건강에 흥미없는 당신에게, 건강 발랄∼？세트
CASE 5
	PRINTL 「축하합니다, 5등이에요」
	PRINTL 「체력 부족해 고민하시는 당신에게, 건강 발랄∼？세트를 드립니다」
	PRINTW 「피로가 심할 때는 이것으로 극복해 주세요」
	PRINTL 내용은 정력제, 불끈불끈Z, 포션, 우콘드링크 5개
	PRINTL 녹즙 과립 10배분 (모리야의 맛있는 물 2 L 물병(와인) 부착), 스포츠드링크 10개였다
	PRINTW 가지고 걷는 것만으로 지칩니다만……
	ITEM:정력제 = MIN(ITEM:정력제 + 1, 99)
	ITEM:불끈불끈Z = MIN(ITEM:불끈불끈Z + 1, 99)
	ITEM:포션 = MIN(ITEM:포션 + 1, 99)
	ITEM:우콘드링크 = MIN(ITEM:우콘드링크 + 5, 99)
	ITEM:녹즙 = MIN(ITEM:녹즙 + 10, 99)
	ITEM:스포츠드링크 = MIN(ITEM:스포츠드링크 + 10, 99)
	PRINTW 「아, 가지지 못했던 만큼은 모리야 신사에의 기증 취급으로 반환해 주십니다」

;6등：이것으로 그 아이와 이이사이에!사랑의 해프닝 세트
CASE 6
	PRINTL 「6등이 됩니다」
	PRINTL 「경품은… 사랑의 해프닝 세트, 라고 합니다」
	PRINTW 얼굴을 붉히면서 상품을 내 왔다.
	ITEM:로션 = MIN(ITEM:로션 + 1, 99)
	ITEM:미약 = MIN(ITEM:미약 + 1, 99)
	ITEM:춘약 = MIN(ITEM:춘약 + 1, 99)
	ITEM:콘돔 = MIN(ITEM:콘돔 + 1, 99)
	ITEM:피임약 = MIN(ITEM:피임약 + 1, 99)

;7등：소득의 기쁨을 당신에게, 야채의모종 세트(야채가 생긴다고는 말하지 않았다)
CASE 7
	PRINTL 「7등입니다」
	PRINTL 「상품은 야채의모종 세트입니다」
	PRINTL 「노력해 풍작 목표로 해 주세요」
	PRINTW 「덧붙여 우리 모리야 신사에서는 풍작 기원 및 봉납의 접수도 가고 있습니다」
	ITEM:양배추의모종 = MIN(ITEM:양배추의모종 + 1, 99)
	ITEM:딸기의모종 = MIN(ITEM:딸기의모종 + 1, 99)
	ITEM:순무의모종 = MIN(ITEM:순무의모종 + 1, 99)
	ITEM:오이의모종 = MIN(ITEM:오이의모종 + 1, 99)
	ITEM:토마토의모종 = MIN(ITEM:토마토의모종 + 1, 99)
	ITEM:수박의모종 = MIN(ITEM:수박의모종 + 1, 99)
	ITEM:가지의모종 = MIN(ITEM:가지의모종 + 1, 99)
	ITEM:고구마의모종 = MIN(ITEM:고구마의모종 + 1, 99)
	ITEM:유리의모종 = MIN(ITEM:유리의모종 + 1, 99)
	ITEM:당근의모종 = MIN(ITEM:당근의모종 + 1, 99)
	ITEM:파의모종 = MIN(ITEM:파의모종 + 1, 99)
	ITEM:로터의모종 = MIN(ITEM:로터의모종 + 1, 99)

	PRINTW 「아, 가지지 못했던 만큼은 모리야 신사에의 기증 취급으로 돌려주세요」

;8등：한 번 먹으면 멈추지 않는, 의리 초콜렛의 임금님의 엔트리다!
CASE 8
	PRINTL 「8등입니다」
	PRINTL 「1개 먹으면 그만둘 수 없다 멈출 수 없다, 의리 초콜렛의 대명사의 엔트리다!」
	PRINTL 「이라고 하는 것으로 이쪽을 부디」
	PRINTW 그렇게 말하면서 건네받은 것은 초콜렛 과자의 상자였다
	ITEM:블랙썬더 = MIN(ITEM:블랙썬더 + 20, 99)

;유감… 9등：수세미, 10개 모으면 금 수세미와 교환입니다!
CASE 9
	PRINTL 「이것은 특별상이군요」
	PRINTL 「특별상은 수세미입니다」
	PRINTL 「이 수세미, 단순한 수세미가 아닙니다」
	PRINTW 「10개 모아 봉납하면, 놀랍게도 저쪽의 금 수세미와 교환이 됩니다」
	FLAG:999 += 1 << 5

	IF (FLAG:999 >> 5 & 15) == 10
		PRINTL 「축하합니다, 수세미를 10개 모았기에 금 수세미와 교환하겠습니다」
		PRINTL 금 수세미를 손에 넣었다!
		PRINTFORMW \@GETBIT(FLAG:999, 9)? 청소 # 요리 \@가 조금 잘 되는 듯한 느낌이다
		IF GETBIT(FLAG:999, 9)
			EXP:MASTER:청소경험 += 1
		ELSE
			EXP:MASTER:요리경험 += 1
		ENDIF
		;금 수세미의 효과를 변환, 수세미 수집 개수를 리셋
		;15(0b1111)를 5 비트 쉬프트 해, 비트 반전한 것(0b000011111)를 FLAG:999로 비트마다의 논리합 시키고 있습니다, 아마
		INVERTBIT FLAG:999, 9
		FLAG:999 &= ~(15 << 5)
	ELSE
		PRINTFORMW 「현재 수세미는 {(FLAG:999 >> 5) & 15}개 모였습니다. 금 수세미를 목표로 힘내주세요」
	ENDIF
;스카：모리야부적(꽝 복권), 5장으로 다시 한번 파하는 돈
CASE 10
	PRINTW 상냥한 웃는 얼굴로 복권을 당신에게 건네주어 왔다
	PRINTW ……
	PRINTW …
	PRINTL ┌─────────────┐
	PRINTL ｜　　　　　　　　　　　　　　　｜
	PRINTL ｜　　모리야 부적 　　　　　　　｜
	PRINTL ｜　　　　　　　　　　　　　　　｜
	PRINTL ｜　　5장 모으면　　　　　　　　｜
	PRINTL ｜　　　복권을 1회 뽑기 가능 　｜
	PRINTL ｜　　　　　　　　　　　　　　　｜
	PRINTL ｜　이익은 보증하기 어렵습니다　｜
	PRINTL ｜　　　　　　　　　　　　　　　｜
	PRINTL ｜　　　　　　　　모리야 신사 　｜
	PRINTL └─────────────┘
	PRINTW 에둘러 말해서 꽝이잖아. 이거 허탕 아냐?
	PRINTFORMW 「다음엔 꼭 좋은 걸 뽑을 수 있어요. 힘내라%UNICODE(0x2665)% 힘내라%UNICODE(0x2665)%」
	ITEM:모리야부적 ++

CASEELSE
	PRINTFORML 「어라, 이상하네요」
	PRINTFORML 복권의 내용을 봐 고개를 갸웃하고 있다
	PRINTFORMW 「이런 복권 안 들어있을 텐데요」
	PRINTFORML 「어쩔 수 없죠. 이번엔 없던 걸로 치고 대금은 돌려드릴게요」
	PRINTFORMW 아무래도 관계없는 복권이 섞여있던 모양이다
	MONEY += 3000
ENDSELECT
RETURN 0

@모리야복권데이트상대의리셋()
;모리야복권의 데이트 때에 해당 캐릭터의 기상시간 etc를 바탕으로 되돌린다
IF FLAG:모리야복권데이트상대
	CFLAG:(FLAG:모리야복권데이트상대):내방시간 = CSVCFLAG(FLAG:모리야복권데이트상대, 352)
	CFLAG:(FLAG:모리야복권데이트상대):귀가시간 = CSVCFLAG(FLAG:모리야복권데이트상대, 353)
	CFLAG:(FLAG:모리야복권데이트상대):취침시간 = CSVCFLAG(FLAG:모리야복권데이트상대, 354)
	CFLAG:(FLAG:모리야복권데이트상대):기상시간 = CSVCFLAG(FLAG:모리야복권데이트상대, 355)
	FLAG:모리야복권데이트상대 = 0
ENDIF


code:
@봇슈트
;팬티몰수, 수집

;SIF !CFLAG:TARGET:팬티2 && STRLENS(CSTR:MASTER:12)
SIF !CFLAG:TARGET:팬티2 && FLAG:팬티수송
	CALL 팬티몰수(TARGET)
;요리 가지고 있으면(자) 먹어진다
IF  DISHTYPE
	SIF DISHASSI == TARGET
		RETURN 0
	SIF CFLAG:TARGET:동행
		RETURN 0
	SIF DISH_DRAG()
		RETURN 0
	LOCAL:1 = 0
	IF DISHTYPE == 3
		SIF TIME_PROGRESS(TCVAR:디저트공복시각)
			LOCAL:1 = 1
	ELSE
		SIF TIME_PROGRESS(TCVAR:TARGET:공복시각)
			LOCAL:1 = 1
	ENDIF
	IF FLAG:20 && LOCAL:1 && RAND:5 == 0 && SELECTCOM != 413
		PRINTL
		PRINTFORMW %타겟은% %마스터가% 가지고 있던＜%EX_DISHNAME()%＞를 찾아냈다
		PRINTFORMW
		PRINTFORMW
		PRINTFORMW 만든 요리를 먹을 수 있어 버렸다
		SIF DISH_RADICAL()
			CALL MESSAGE_REACTION_TASTE
		CALL KOJO_MESSAGE_SEND("EVENT", 14, TARGET,-1,-1)
		CALL 식사(TARGET)
		CALL RESET_DISH
	ENDIF
ENDIF

@팬티몰수(ARG)
;팬티몰수, 수집
LOCAL:1 = RAND:2
SIF FLAG:발원내용 == 202
	LOCAL:1 = RAND:3
IF  !LOCAL:1 && CFLAG:ARG:노팬티 != 2
	PRINTL
	PRINTFORMW %CALLNAME:ARG%에 보고 나무랄 수 있는 신체 검사를 받았다…
	PRINTFORMW
	PRINTFORMW
	PRINTFORMW
	IF TALENT:MASTER:팬티밀수 && RAND:10
		PRINTFORMW 그러나 %마스터는% 교묘하게 %CALLNAME:ARG%의 눈을 속였다
	ELSE
		PRINTFORMW 훔친 팬티를 몰수당했다
		SIF !CFLAG:약점잡음
			PRINTFORMW %CALLNAME:ARG%에게 약점을 잡혔다
			CFLAG:ARG:약점잡음 += 5
		CALL KOJO_MESSAGE_SEND("EVENT", 9, ARG, 0,1)
		;CSTR:MASTER:12 =
		VARSET PANTS_TEMP
		FLAG:팬티수송 = 0
	ENDIF
ENDIF


code:
;-------------------------------------------------
;贈り物を生成する関数
;RESULT:0  GIFT_ID
;RESULT:1  値段
;RESULTS   名前
;-------------------------------------------------
@CREATE_GIFT(AREA)
#DIM AREA
#DIM MAIN
#DIM ADJECT
#DIM BASIC_PRICE
#DIM MAGNIF_PRICE
#DIMS GIFTNAME_MAIN
#DIMS GIFTNAME_ADJECT
VARSET RESULT

CALL DRAW_GIFT_MAIN(AREA)
MAIN = RESULT:0
BASIC_PRICE = RESULT:1
GIFTNAME_MAIN = %RESULTS:0%

CALL DRAW_GIFT_ADJECT(AREA, RESULTS:1)
ADJECT = RESULT:0
MAGNIF_PRICE = RESULT:1
GIFTNAME_ADJECT = %RESULTS:0%

;値段の設定
RESULT:1 = BASIC_PRICE * MAGNIF_PRICE / 100
;半額セールとか値引き要素等
SIF 0
	TIMES RESULT:1, 0.50

;名前の設定
RESULTS = %GIFTNAME_ADJECT%%GIFTNAME_MAIN%
;IDの設定
RESULT:0 = ADJECT * 1000 + MAIN
RETURN RESULT:0

;-------------------------------------------------
;贈り物본체を抽選する関数
;-------------------------------------------------
@DRAW_GIFT_MAIN(AREA)
#DIM AREA
#DIM CATEGORY_MAIN
#DIM GIFT_MAIN
#DIM 카테고리상한
VARSET RESULT
카테고리상한 = 52
WHILE !RESULT
	CATEGORY_MAIN = RAND:카테고리상한 + 1
	GIFT_MAIN = CATEGORY_MAIN * 10 + RAND:10
	CALL GIFTDATA_MAIN(GIFT_MAIN, AREA)
WEND
RETURN GIFT_MAIN

;-------------------------------------------------
;贈り物の형용사を抽選する関数
;-------------------------------------------------
@DRAW_GIFT_ADJECT(AREA, SENSE)
#DIM AREA
#DIM CATEGORY_ADJECT
#DIM GIFT_ADJECT
#DIMS SENSE
#DIM 카테고리상한
VARSET RESULT
카테고리상한 = 13
WHILE !RESULT
	CATEGORY_ADJECT = RAND:카테고리상한 + 1
	GIFT_ADJECT = CATEGORY_ADJECT * 100 + RAND:100
	CALL GIFTDATA_ADJECT(GIFT_ADJECT, AREA, SENSE)
WEND
RETURN GIFT_ADJECT

;-------------------------------------------------
;贈り物を평가する関数
;-------------------------------------------------
@REVIEW_GIFT(CHARA, GIFT_ID)
#DIM CHARA
#DIM GIFT_ID
#DIMS SENSE_CHARA_GOOD
#DIMS SENSE_CHARA_BAD
#DIMS SENSE_GIFT
#DIM SENSE_GOOD
#DIM SENSE_BAD
#DIM MAGNIF_PRICE
#DIM BASIC_PRICE

;SENSE抽出
SENSE_GIFT =
CALL GIFTDATA_ADJECT(GET_GIFTDATA(GIFT_ID, "형용사"))
SENSE_GIFT += RESULTS:1
MAGNIF_PRICE = RESULT:1

CALL GIFTDATA_MAIN(GET_GIFTDATA(GIFT_ID, "본체"))
SENSE_GIFT += RESULTS:1
BASIC_PRICE = RESULT:1

;정가を見て10,000以上なら고급、1,500未満なら싸구려
IF BASIC_PRICE * MAGNIF_PRICE / 100 >= 10000
	SENSE_GIFT += "고급/"
ELSEIF BASIC_PRICE * MAGNIF_PRICE / 100 < 1500
	SENSE_GIFT += "싸구려/"
ENDIF

;캐릭터の感性
SENSE_CHARA_GOOD = %GET_STR(, "캐릭터데이터", CHARA, "감성：좋아함")%
SENSE_CHARA_BAD = %GET_STR(, "캐릭터데이터", CHARA, "감성：싫어함")%

;共通設定：好き
;［유치］は캐릭터が好き
SIF TALENT:CHARA:유치
	SENSE_CHARA_GOOD += "캐릭터/" * 3
;［요정］は자연が好き
SIF TALENT:CHARA:요정
	SENSE_CHARA_GOOD += "자연/"
;［츠쿠모가미］は츠쿠모가미と도구が好き
SIF TALENT:CHARA:츠쿠모가미
	SENSE_CHARA_GOOD += "츠쿠모가미/도구/"
;요괴は요괴が好き
SIF 요괴체커(CHARA)
	SENSE_CHARA_GOOD += "요괴/"
;CSVの전투능력があるなら스펠카드は好き
SIF CSVABL(CHARA, GETNUM(ABL, "전투능력")) > 0
	SENSE_CHARA_GOOD += "스펠카드/" * 3

;共通設定：嫌い
;조악はみんな嫌い
SIF 1
	SENSE_CHARA_BAD += "조악/" * 5
;요괴は주술が嫌い
IF 요괴체커(CHARA)
	SENSE_CHARA_BAD += "주술/" * 3
	;법구が好きでないなら법구は嫌い
	SIF STRFIND(SENSE_CHARA_GOOD, "법구/") < 0
		SENSE_CHARA_BAD += "법구/" * 3
ENDIF
;요괴ではなく요술が好きでないなら요술は嫌い
IF !요괴체커(CHARA) && STRFIND(SENSE_CHARA_GOOD, "요술/") < 0
	SENSE_CHARA_BAD += "요술/" * 3
	;さらに뼈、피、독も同様
	SIF STRFIND(SENSE_CHARA_GOOD, "뼈/") < 0
		SENSE_CHARA_BAD += "뼈/" * 3
	SIF STRFIND(SENSE_CHARA_GOOD, "피/") < 0
		SENSE_CHARA_BAD += "피/" * 3
	SIF STRFIND(SENSE_CHARA_GOOD, "독/") < 0
		SENSE_CHARA_BAD += "독/" * 3
ENDIF
;술내성が下戸なら술は嫌い
IF TALENT:CHARA:술내성 < -1
	SENSE_CHARA_BAD += "술/" * 3
ELSEIF TALENT:CHARA:술내성 < 0
	SENSE_CHARA_BAD += "술/"
ENDIF
;CSVの전투능력がないなら스펠카드はちょっと…
SIF CSVABL(CHARA, GETNUM(ABL, "전투능력")) <= 0
	SENSE_CHARA_BAD += "스펠카드/"

;평가득점
SENSE_GOOD = 300 + 100 * STR_MULTI_COUNT(SENSE_CHARA_GOOD, SENSE_GIFT)

;;무드による補正,※封印しました
;IF BASE:CHARA:무드 >= MAXBASE:CHARA:무드
;	TIMES SENSE_GOOD, 1.00
;ELSEIF BASE:CHARA:무드 >= MAXBASE:CHARA:무드 / 2
;	TIMES SENSE_GOOD, 0.80
;ELSEIF BASE:CHARA:무드 >= MAXBASE:CHARA:무드 / 4
;	TIMES SENSE_GOOD, 0.60
;ELSE
;	TIMES SENSE_GOOD, 0.40
;ENDIF

;;自宅地域による補正,※封印しました
;SIF CFLAG:CHARA:데이트중  == CFLAG:CHARA:자택위치 / 100
;	TIMES SENSE_GOOD, 0.80

;祭日による補正
SIF FESTIVAL() != ""
	TIMES SENSE_GOOD, 1.25

;평가減点
SENSE_BAD = 300 + 100 * STR_MULTI_COUNT(SENSE_CHARA_BAD, SENSE_GIFT)

;総評
RESULT = SENSE_GOOD * 1000 / (SENSE_GOOD + SENSE_BAD)
[IF_DEBUG]
	PRINTFORML 좋아함：%SENSE_CHARA_GOOD%
	PRINTFORML 싫어함：%SENSE_CHARA_BAD%
	PRINTFORML 물건：%SENSE_GIFT%
	PRINTFORML 득점：{SENSE_GOOD}
	PRINTFORML 감점：{SENSE_BAD}
	PRINTFORML 결과：{RESULT}
	WAIT
[ENDIF]
RETURN RESULT
;-------------------------------------------------
;요괴かどうか判定する関数
;COMMONのがいいかも
;-------------------------------------------------
@요괴체커(ARG)
#FUNCTION
SIF TALENT:ARG:인간
	RETURNF 0
SIF TALENT:ARG:신령
	RETURNF 0
SIF TALENT:ARG:요정
	RETURNF 0
SIF TALENT:ARG:봉래인
	RETURNF 0
RETURNF 1

;-------------------------------------------------
;一日の終わりに贈り物をお気に入りにするかの関数
;-------------------------------------------------
@GIFT_FAVORITE
#DIM CHARA
#DIM GIFT_ID
#DIM GIFT_ADJECT
#DIM GIFT_MAIN
#DIM GIFT_SCORE, 3
#DIMS GIFT_NAME
#DIMS SENSE
FOR CHARA, 1, CHARANUM
	SIF !TCVAR:CHARA:오늘의선물
		CONTINUE

	GIFT_ID = TCVAR:CHARA:오늘의선물
	GIFT_ADJECT = GET_GIFTDATA(GIFT_ID, "형용사")
	GIFT_MAIN = GET_GIFTDATA(GIFT_ID, "본체")
	GIFT_SCORE:1 = GET_GIFTDATA(GIFT_ID, "득점")

	;贈り物データの参照
	CALL GIFTDATA_ADJECT(GIFT_ADJECT)
	LOCALS:1 = %RESULTS:0%
	CALL GIFTDATA_MAIN(GIFT_MAIN)
	SENSE = %RESULTS:1%
	LOCALS:2 = %RESULTS:0%
	GIFT_NAME = %LOCALS:1%%LOCALS:2%
;	PRINTFORMW SCORE{GIFT_SCORE:1}

	PRINTL
	PRINTFORM %조사처리(CALLNAME:CHARA,"는")% %CALLNAME:MASTER%에게 받은 %GIFT_NAME%
	IF GIFT_SCORE:1 >= 700
		IF STRFIND(SENSE, "움켜쥠") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 소중하게 움켜쥐고 잠들었다……。
		ELSEIF STRFIND(SENSE, "품음") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 생각하며 끌어안았다……。
		ELSEIF STRFIND(SENSE, "몸에걸침") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 시험삼아 몸에 걸치고, 거울 앞에서 얼굴을 붉혔다……。
		ELSE
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 넋을 잃고 바라보다가 잠이 들었다……。
		ENDIF
		;現在の贈り物と比較してお気に入りにする
		IF CFLAG:CHARA:받은선물
			GIFT_SCORE:2 = GET_GIFTDATA(CFLAG:CHARA:받은선물, "득점")
		ELSE
			GIFT_SCORE:2 = 0
		ENDIF
		IF GIFT_SCORE:1 >= GIFT_SCORE:2
			SIF CFLAG:CHARA:받은선물
				CFLAG:CHARA:선물스톡 = CFLAG:CHARA:받은선물
			CFLAG:CHARA:받은선물 = TCVAR:CHARA:오늘의선물
			CALL COLORMESSAGE(@"%조사처리(GIFT_NAME,"는")% %CALLNAME:CHARA%의 마음에 들었다!", C_YELLOW, 2)
		ENDIF
	ELSEIF GIFT_SCORE:1 >= 400
		IF STRFIND(SENSE, "움켜쥠") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 선반 위에 놓았다……。
		ELSEIF STRFIND(SENSE, "품음") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 책상 위에 놓았다……。
		ELSEIF STRFIND(SENSE, "몸에걸침") >= 0
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 옷장에 넣었다……。
		ELSE
			PRINTFORMW %조사만처리(GIFT_NAME,"를")% 일단 보관했다……。
		ENDIF
	ELSE
		PRINTFORMW %조사만처리(GIFT_NAME,"를")% 슬쩍 저당잡았다……。
	ENDIF
	PRINTL
NEXT

;-------------------------------------------------
;Look内で받은선물を表示する関数
;-------------------------------------------------
@LOOK_GIFT(CHARA)
#DIM CHARA
#DIM GIFT_NAME
#DIM GIFT_SCORE
;우후후中は表示しない
SIF CFLAG:CHARA:우후후
	RETURN

;お気に入り
IF CFLAG:CHARA:받은선물
	PRINTL
	PRINT 【마음에 듦】
	CALL COLORMESSAGE("★", C_YELLOW, 0)
	CALL GIFTNAME(CFLAG:CHARA:받은선물)
	PRINTFORM %RESULTS%
	[IF_DEBUG]
		CALL PRINT_GIFTDATA(CFLAG:CHARA:받은선물, CHARA)
	[ENDIF]
;お気に入りでない場合
ELSEIF CFLAG:CHARA:선물스톡
	GIFT_SCORE = GET_GIFTDATA(CFLAG:CHARA:선물스톡, "득점")
	;400未満は휴대しない
	IF GIFT_SCORE < 400
		RETURN
	ELSE
		;평가득점が高いほど휴대しやすい
		SIF FLAG:일일이벤트 * 10 > GIFT_SCORE
			RETURN
	ENDIF
	PRINTL
	PRINT 【 휴대 】
	CALL GIFTNAME(CFLAG:CHARA:선물스톡)
	PRINTFORM %RESULTS%
ENDIF

;-------------------------------------------------
;贈り物データを表示する関数
;-------------------------------------------------
@PRINT_GIFTDATA(GIFT_ID, CHARA)
#DIM GIFT_ID
#DIM CHARA
#DIM 횟수
#DIM GIFT_SOCRE
#DIM 받은날
#DIM 산장소
#DIM GIFT_ADJECT
#DIM GIFT_MAIN
#DIMS GIFT_NAME
#DIM MAGNIF_PRICE
#DIM SELL_AREA_ADJECT
#DIM BASIC_PRICE
#DIM SELL_AREA_MAIN
#DIMS SENSE

횟수 = GET_GIFTDATA(GIFT_ID, "횟수")
GIFT_SOCRE = GET_GIFTDATA(GIFT_ID, "득점")
받은날 = GET_GIFTDATA(GIFT_ID, "날짜")
산장소 = GET_GIFTDATA(GIFT_ID, "장소")
GIFT_ADJECT = GET_GIFTDATA(GIFT_ID, "형용사")
GIFT_MAIN = GET_GIFTDATA(GIFT_ID, "본체")
CALL GIFTDATA_ADJECT(GIFT_ADJECT)
MAGNIF_PRICE = RESULT:1
SELL_AREA_ADJECT = RESULT:2
LOCALS:1 = %RESULTS:0%
SENSE = %RESULTS:1%
CALL GIFTDATA_MAIN(GIFT_MAIN)
BASIC_PRICE = RESULT:1
SELL_AREA_MAIN = RESULT:2
LOCALS:2 = %RESULTS:0%
GIFT_NAME = %LOCALS:1%%LOCALS:2%
SENSE += RESULTS:1

[IF_DEBUG]
	PRINTL
	CALL PRINT_DATE(받은날)
	PRINTFORML %GET_MAPNAME(산장소)%에서 %CALLNAME:CHARA%에게 {횟수}번째로 사준 %GIFT_NAME%
	PRINTFORML 정가는 {BASIC_PRICE} * {MAGNIF_PRICE}\% = \\{BASIC_PRICE * MAGNIF_PRICE / 100}
	PRINTFORML 평가득점은　{GIFT_SOCRE }점
	PRINTFORM 이 형용사를 파는 곳은
	SIF GETBIT(SELL_AREA_ADJECT, 0)
		PRINT 　신사
	SIF GETBIT(SELL_AREA_ADJECT, 1)
		PRINT 　절
	SIF GETBIT(SELL_AREA_ADJECT, 2)
		PRINT 　마을
	SIF GETBIT(SELL_AREA_ADJECT, 3)
		PRINT 　호수
	SIF GETBIT(SELL_AREA_ADJECT, 4)
		PRINT 　죽림
	SIF GETBIT(SELL_AREA_ADJECT, 5)
		PRINT 　숲
	SIF GETBIT(SELL_AREA_ADJECT, 6)
		PRINT 　명계
	SIF GETBIT(SELL_AREA_ADJECT, 7)
		PRINT 　산록
	SIF GETBIT(SELL_AREA_ADJECT, 8)
		PRINT 　산정
	SIF GETBIT(SELL_AREA_ADJECT, 9)
		PRINT 　지저
	SIF GETBIT(SELL_AREA_ADJECT, 10)
		PRINT 　달
	PRINTL
	PRINTFORM 이 본체를 파는 곳은
	SIF GETBIT(SELL_AREA_MAIN, 0)
		PRINT 　신사
	SIF GETBIT(SELL_AREA_MAIN, 1)
		PRINT 　절
	SIF GETBIT(SELL_AREA_MAIN, 2)
		PRINT 　마을
	SIF GETBIT(SELL_AREA_MAIN, 3)
		PRINT 　호수
	SIF GETBIT(SELL_AREA_MAIN, 4)
		PRINT 　죽림
	SIF GETBIT(SELL_AREA_MAIN, 5)
		PRINT 　숲
	SIF GETBIT(SELL_AREA_MAIN, 6)
		PRINT 　명계
	SIF GETBIT(SELL_AREA_MAIN, 7)
		PRINT 　산록
	SIF GETBIT(SELL_AREA_MAIN, 8)
		PRINT 　산정
	SIF GETBIT(SELL_AREA_MAIN, 9)
		PRINT 　지저
	SIF GETBIT(SELL_AREA_MAIN, 10)
		PRINT 　달
	PRINTL
	IF STRFIND(SENSE, "움켜쥠") >= 0
		PRINTFORML 움켜쥘 수 있는 것이다
	ELSEIF STRFIND(SENSE, "품음") >= 0
		PRINTFORML 껴안을 수 있는 것이다
	ELSEIF STRFIND(SENSE, "몸에걸침") >= 0
		PRINTFORML 몸에 걸치는 것이다
	ELSE
		PRINTFORML 바라보는 것이다
	ENDIF
	PRINTFORML SENSE：%SENSE%
	RETURN
[ENDIF]

;-------------------------------------------------
;贈り物データを抽出する関数
;-------------------------------------------------
@GET_GIFTDATA(GIFT_ID, ARGS)
#FUNCTION
#DIM GIFT_ID
SELECTCASE ARGS
CASE "횟수"
	RETURNF GIFT_ID / 10000000000000000
CASE "득점"
	RETURNF GIFT_ID / 10000000000000 % 1000
CASE "날짜"
	RETURNF GIFT_ID / 1000000000 % 10000
CASE "장소"
	RETURNF GIFT_ID / 10000000 % 100
CASE "형용사"
	RETURNF GIFT_ID / 1000 % 10000
CASE "본체"
	RETURNF GIFT_ID % 1000
ENDSELECT

;-------------------------------------------------
;贈り物の名前を出力する関数
;FUNCTIONSにしたかったけどCALLできなくて泣く泣くこのカタチに…
;-------------------------------------------------
@GIFTNAME(GIFT_ID)
#DIM GIFT_ID
#DIM GIFT_ADJECT
#DIM GIFT_MAIN
GIFT_ADJECT = GET_GIFTDATA(GIFT_ID, "형용사")
GIFT_MAIN = GET_GIFTDATA(GIFT_ID, "본체")

CALL GIFTDATA_ADJECT(GIFT_ADJECT)
LOCALS:1 = %RESULTS:0%
CALL GIFTDATA_MAIN(GIFT_MAIN)
LOCALS:2 = %RESULTS:0%

RESULTS = %LOCALS:1%%LOCALS:2%
RETURN

;-------------------------------------------------
;引数に地域を羅列するとそこのbitを入れて返す関数Ver.2.0
;"전부"を入れると전부入れた数値(2^11-1)を返す
;"以外"を入れると逆になる
;COMMONに入れてもいいかもしれない（どこで使うか知らん）
;-------------------------------------------------
@AREA_SETBIT(ARGS)
#FUNCTION
VARSET RESULT
SIF STRFIND(ARGS, "전부") >= 0
	RETURNF 2047

SIF STRFIND(ARGS, "신사") >= 0
	SETBIT RESULT, 0
SIF STRFIND(ARGS, "절") >= 0
	SETBIT RESULT, 1
SIF STRFIND(ARGS, "마을") >= 0
	SETBIT RESULT, 2
SIF STRFIND(ARGS, "호수") >= 0
	SETBIT RESULT, 3
SIF STRFIND(ARGS, "죽림") >= 0
	SETBIT RESULT, 4
SIF STRFIND(ARGS, "숲") >= 0
	SETBIT RESULT, 5
SIF STRFIND(ARGS, "명계") >= 0
	SETBIT RESULT, 6
SIF STRFIND(ARGS, "산록") >= 0
	SETBIT RESULT, 7
SIF STRFIND(ARGS, "산정") >= 0
	SETBIT RESULT, 8
SIF STRFIND(ARGS, "지저") >= 0
	SETBIT RESULT, 9
SIF STRFIND(ARGS, "달") >= 0
	SETBIT RESULT, 10

IF STRFIND(ARGS, "이외") >= 0
	FOR LOCAL, 0, 11
		INVERTBIT RESULT, LOCAL
	NEXT
ENDIF
RETURNF RESULT

path: ERB/이벤트관련/선물관련/선물.ERB